

This worksheet is meant as a complement to “Computing the Chow variety of quadratic space curves”, by Peter Bürgisser, Kathlén Kohn, Pierre Lairez, and Bernd Sturmfels.

The package FGb, to compute efficiently Groebner basis, is available at <http://www-polsys.lip6.fr/~jcf/Software/FGb/Download/index.html>

```
> with(FGb): with(LinearAlgebra):
```

The computation of some of the Groebner bases require ~10 GB of RAM, so for convenienc, some of the results are preloaded.

```
> read "GBs.mpl":
```

Section 1] Coordinates and quadrics in G(2,4)

Stiefel coordinates: a point in G(2,4) is given as the column space of a 4x2 matrix.

```
> stiefelMat := Matrix(4,2,(i,j)->a[i+10*j]);
```

$$\text{stiefelMat} := \begin{bmatrix} a_{11} & a_{21} \\ a_{12} & a_{22} \\ a_{13} & a_{23} \\ a_{14} & a_{24} \end{bmatrix} \quad (1.1)$$

```
> stiefelVars := [op(indets(stiefelMat))];
```

$$\text{stiefelVars} := [a_{11}, a_{12}, a_{13}, a_{14}, a_{21}, a_{22}, a_{23}, a_{24}] \quad (1.2)$$

(Dual) Plucker coordinates: a point in G(2,4) is given by the 2x2 minors of the matrix of Stiefel coordinates.

```
> pluckerCo := [seq(seq(p[i-1,j-1]=Determinant(stiefelMat[[i,j],1..2]), j=i+1..4), i=1..3)];
```

$$\text{pluckerCo} := [p_{0,1} = a_{11}a_{22} - a_{12}a_{21}, p_{0,2} = a_{11}a_{23} - a_{13}a_{21}, p_{0,3} = a_{11}a_{24} - a_{14}a_{21}, p_{1,2} = a_{12}a_{23} - a_{13}a_{22}, p_{1,3} = a_{12}a_{24} - a_{14}a_{22}, p_{2,3} = a_{13}a_{24} - a_{14}a_{23}] \quad (1.3)$$

```
> pluckerDual := { p[0,1] = p[2,3], p[0,2] = -p[1,3], p[0,3] = p[1,2], p[1,2] = p[0,3], p[1,3] = -p[0,2], p[2,3] = p[0,1] };
```

$$\text{pluckerDual} := \{p_{0,1} = p_{2,3}, p_{0,2} = -p_{1,3}, p_{0,3} = p_{1,2}, p_{1,2} = p_{0,3}, p_{1,3} = -p_{0,2}, p_{2,3} = p_{0,1}\} \quad (1.4)$$

```
> pluckerVars := convert(select(has(indets(pluckerCo),p),list);
```

$$\text{pluckerVars} := [p_{0,1}, p_{0,2}, p_{0,3}, p_{1,2}, p_{1,3}, p_{2,3}] \quad (1.5)$$

Plucker coordinates are bound by the Plucker relation

```
> pluckerRel := p[0,1]*p[2,3]-p[0,2]*p[1,3]+p[0,3]*p[1,2];
```

$$\text{pluckerRel} := p_{0,1}p_{2,3} - p_{0,2}p_{1,3} + p_{0,3}p_{1,2} \quad (1.6)$$

A generic 6x6 symmetric matrix. The change of basis $c[5]=c[5]+c[12]$, $c[9]=c[9]-c[12]$ makes $c[12]$ disappear below.

```
> sym6 := subs([c[5]=c[5]+c[12], c[9]=c[9]-c[12]], Matrix(6, 6, [c
[0], c[1], c[2], c[3], c[4], c[5], c[1], c[6], c[7], c[8], c[9],
c[10], c[2], c[7], c[11], c[12], c[13], c[14], c[3], c[8], c[12],
c[15], c[16], c[17], c[4], c[9], c[13], c[16], c[18], c[19], c
[5], c[10], c[14], c[17], c[19], c[20]]));
```

$$\text{sym6} := \begin{bmatrix} c_0 & c_1 & c_2 & c_3 & c_4 & c_5 + c_{12} \\ c_1 & c_6 & c_7 & c_8 & c_9 - c_{12} & c_{10} \\ c_2 & c_7 & c_{11} & c_{12} & c_{13} & c_{14} \\ c_3 & c_8 & c_{12} & c_{15} & c_{16} & c_{17} \\ c_4 & c_9 - c_{12} & c_{13} & c_{16} & c_{18} & c_{19} \\ c_5 + c_{12} & c_{10} & c_{14} & c_{17} & c_{19} & c_{20} \end{bmatrix} \quad (1.7)$$

```
> cvars := [seq(c[i], i=0..20)];
cvars := [c_0, c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_9, c_{10}, c_{11}, c_{12}, c_{13}, c_{14}, c_{15}, c_{16}, c_{17}, c_{18}, c_{19}, c_{20}] \quad (1.8)
```

The ideals we will compute do not depend on $c[12]$, as explain in the article.

```
> cvars0 := remove('=', cvars, c[12]);
cvars0 := [c_0, c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_9, c_{10}, c_{11}, c_{13}, c_{14}, c_{15}, c_{16}, c_{17}, c_{18}, c_{19}, c_{20}] \quad (1.9)
```

A generic quadratic form in the Plucker coordinates.

```
> gQ := collect(Vector[row](pluckerVars).sym6.Vector(pluckerVars),
pluckerVars, distributed);
gQ := c_0 p_{0,1}^2 + 2 c_4 p_{1,3} p_{0,1} + (2 c_5 + 2 c_{12}) p_{2,3} p_{0,1} + 2 c_1 p_{0,2} p_{0,1}
+ 2 c_2 p_{0,3} p_{0,1} + 2 c_3 p_{1,2} p_{0,1} + c_6 p_{0,2}^2 + (2 c_9 - 2 c_{12}) p_{1,3} p_{0,2}
+ 2 c_{10} p_{2,3} p_{0,2} + 2 c_7 p_{0,3} p_{0,2} + 2 c_8 p_{1,2} p_{0,2} + c_{11} p_{0,3}^2 + 2 c_{12} p_{1,2} p_{0,3}
+ 2 c_{13} p_{1,3} p_{0,3} + 2 c_{14} p_{2,3} p_{0,3} + c_{15} p_{1,2}^2 + 2 c_{16} p_{1,3} p_{1,2} + 2 c_{17} p_{2,3} p_{1,2}
+ c_{18} p_{1,3}^2 + 2 p_{2,3} c_{19} p_{1,3} + p_{2,3}^2 c_{20} \quad (1.10)
```

A quadratic form in the Plucker coordinates defines a quadric hypersurface of $G(2,4)$. Two forms that differ by a multiple of the Plucker relation give the same quadric. Thus, the set of quadric of $G(2,4)$ can be identified with the projectivization of the quotient of C^{21} (the quadratic forms) by the Plucker relation (cf. the article).

Section 2] Ideal of coisotropic quadrics in $G(2,4)$

Now, let us see the condition that gQ defines a coisotropic hypersurface.

The quadric defined by gQ is coisotropic if there are $t[0]$, $t[1]$ and $t[2]$, not all zero, such that the following vanishes:

```
> coisoCondition := t[0]*(diff(gQ, p[0,1])*diff(gQ,p[2,3])-diff(gQ,
p[0,2])*diff(gQ,p[1,3])+diff(gQ,p[0,3])*diff(gQ,p[1,2])) + t[1]*
gQ + t[2]*pluckerRel;
```

So here are the equations:

> coisoConditionCoeffs := [coeffs(collect(coisoCondition, pluckerVars, 'distributed'), pluckerVars)];

$$\text{coisoConditionCoeffs} := \left[2 t_1 c_{10} + t_0 (2 (2 c_5 + 2 c_{12}) c_{10} + 4 c_1 c_{20} + 4 c_{14} c_8 + 4 c_7 c_{17} - 2 c_{10} (2 c_9 - 2 c_{12}) - 4 c_6 c_{19}), t_0 (2 (2 c_5 + 2 c_{12}) c_{14} + 4 c_2 c_{20} - 4 c_{10} c_{13} - 4 c_7 c_{19} + 4 c_{14} c_{12} + 4 c_{11} c_{17}) + 2 t_1 c_{14}, t_0 (2 (2 c_5 + 2 c_{12}) c_{17} + 4 c_3 c_{20} + 4 c_{14} c_{15} + 4 c_{12} c_{17} - 4 c_{10} c_{16} - 4 c_8 c_{19}) + 2 t_1 c_{17}, t_0 (2 (2 c_5 + 2 c_{12}) c_{19} + 4 c_4 c_{20} - 4 c_{10} c_{18} - 2 (2 c_9 - 2 c_{12}) c_{19} + 4 c_{13} c_{17} + 4 c_{14} c_{16}) + 2 t_1 c_{19}, t_0 (4 c_4 c_{17} + 4 c_3 c_{19} + 4 c_{13} c_{15} + 4 c_{12} c_{16} - 2 (2 c_9 - 2 c_{12}) c_{16} - 4 c_8 c_{18}) + 2 t_1 c_{16}, t_0 ((2 c_5 + 2 c_{12})^2 + 4 c_0 c_{20} - 4 c_{10} c_4 - 4 c_1 c_{19} + 4 c_{14} c_3 + 4 c_2 c_{17}) + t_1 (2 c_5 + 2 c_{12}) + t_2, t_1 (2 c_9 - 2 c_{12}) - t_2 + t_0 (4 c_{10} c_4 + 4 c_1 c_{19} + 4 c_{13} c_8 + 4 c_7 c_{16} - (2 c_9 - 2 c_{12})^2 - 4 c_6 c_{18}), t_0 (4 c_2 c_{17} + 4 c_3 c_{14} - 4 c_7 c_{16} - 4 c_8 c_{13} + 4 c_{11} c_{15} + 4 c_{12}^2) + t_2 + 2 t_1 c_{12}, t_0 (4 c_4 c_{14} + 4 c_2 c_{19} - 2 (2 c_9 - 2 c_{12}) c_{13} - 4 c_7 c_{18} + 4 c_{13} c_{12} + 4 c_{11} c_{16}) + 2 t_1 c_{13}, 2 t_1 c_8 + t_0 (4 c_3 c_{10} + 4 c_1 c_{17} + 4 c_{12} c_8 + 4 c_7 c_{15} - 2 c_8 (2 c_9 - 2 c_{12}) - 4 c_6 c_{16}), 2 t_1 c_7 + t_0 (4 c_2 c_{10} + 4 c_1 c_{14} + 4 c_{11} c_8 + 4 c_7 c_{12} - 2 c_7 (2 c_9 - 2 c_{12}) - 4 c_6 c_{13}), t_0 (2 c_4 (2 c_5 + 2 c_{12}) + 4 c_0 c_{19} - 2 (2 c_9 - 2 c_{12}) c_4 - 4 c_1 c_{18} + 4 c_{13} c_3 + 4 c_2 c_{16}) + 2 t_1 c_4, t_0 (2 c_3 (2 c_5 + 2 c_{12}) + 4 c_0 c_{17} - 4 c_8 c_4 - 4 c_1 c_{16} + 4 c_{12} c_3 + 4 c_2 c_{15}) + 2 t_1 c_3, t_0 (2 c_2 (2 c_5 + 2 c_{12}) + 4 c_0 c_{14} - 4 c_7 c_4 - 4 c_1 c_{13} + 4 c_{11} c_3 + 4 c_2 c_{12}) + 2 t_1 c_2, t_0 (2 c_1 (2 c_5 + 2 c_{12}) + 4 c_0 c_{10} - 4 c_6 c_4 - 2 c_1 (2 c_9 - 2 c_{12}) + 4 c_7 c_3 + 4 c_2 c_8) + 2 t_1 c_1, t_0 (2 (2 c_5 + 2 c_{12}) c_{20} + 4 c_{14} c_{17} - 4 c_{10} c_{19}) + t_1 c_{20}, t_0 (4 c_3 c_{17} - 4 c_8 c_{16} + 4 c_{12} c_{15}) + t_1 c_{15}, t_0 (-2 (2 c_9 - 2 c_{12}) c_{18} + 4 c_4 c_{19} + 4 c_{13} c_{16}) + t_1 c_{18}, t_0 (4 c_7 c_8 + 4 c_1 c_{10} - 2 c_6 (2 c_9 - 2 c_{12})) + t_1 c_6, t_1 c_{11} + t_0 (4 c_2 c_{14} - 4 c_7 c_{13} + 4 c_{11} c_{12}), t_1 c_0 + t_0 (-4 c_1 c_4 + 2 c_0 (2 c_5 + 2 c_{12}) + 4 c_2 c_3)]$$

The $t[i]$'s are eliminated using linear algebra.

> coisoConditionMat := Matrix(map(e -> [seq(coeff(e, t[i]), i=0..2)], coisoConditionCoeffs));

$$\text{coisoConditionMat} := \left[\begin{array}{l} 21 \times 3 \text{ Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran_order} \end{array} \right] \quad (2.2)$$

The equations we are looking for are the 3x3 minors of the matrix above.

```
> gensCoiso := convert(convert([seq(Determinant(coisoConditionMat
[p,1..3]), p in combinat[choose]([seq(i, i in 1..21]),3))], set),
list):
```

The equations do not depend on c[12].

```
> nops(select(has, gensCoiso, c[12]));
```

0

(2.3)

And we compute a Groebner basis of the ideal generated by these equations. (Preloaded.)

```
> # gbCoiso := fgb_gbasis(gensCoiso, 0, cvars, []):
```

Section 3] Ideal of the Chow forms of two lines

We start with one line given by the Plucker coordinates q[i,j]

The line given by the p[i,j] meets the other if and only if the following vanishes:

```
> meetCondition := p[0,1]*q[2,3] - p[0,2]*q[1,3] + p[0,3]*q[1,2] +
p[2,3]*q[0,1] - p[1,3]*q[0,2] + p[1,2]*q[0,3];
```

$$\text{meetCondition} := p_{0,1} q_{2,3} - p_{0,2} q_{1,3} + p_{0,3} q_{1,2} + p_{1,2} q_{0,3} - p_{1,3} q_{0,2} + p_{2,3} q_{0,1} \quad (3.1)$$

Indeed, translated into Stiefel coordinates, this polynomial is just a determinant:

```
> Matrix([stiefelMat, subs(a=b,stiefelMat)]);
```

$$\begin{bmatrix} a_{11} & a_{21} & b_{11} & b_{21} \\ a_{12} & a_{22} & b_{12} & b_{22} \\ a_{13} & a_{23} & b_{13} & b_{23} \\ a_{14} & a_{24} & b_{14} & b_{24} \end{bmatrix}$$

(3.2)

```
> normal(subs(pluckerCo, subs([p=q,a=b], pluckerCo), meetCondition)
- Determinant(%));
```

0

(3.3)

Therefore, the quadric gQ is the Chow form of a pair of skew lines if it equals the product of form meetCondition above, for some lines q[i,j] and s[i,j], modulo the Plucker relation.

```
> Groebner[Reduce](gQ - meetCondition*subs(q=s, meetCondition),
[pluckerRel], tdeg(op(pluckerVars)));
```

$$\begin{aligned} & (-q_{2,3} s_{2,3} + c_0) p_{0,1}^2 + (-q_{0,1} s_{2,3} + q_{0,3} s_{1,2} + q_{1,2} s_{0,3} - q_{2,3} s_{0,1} \\ & + 2 c_5) p_{2,3} p_{0,1} + (-q_{0,2} s_{1,3} - q_{0,3} s_{1,2} - q_{1,2} s_{0,3} - q_{1,3} s_{0,2} \\ & + 2 c_9) p_{1,3} p_{0,2} + (q_{0,1} s_{1,3} + q_{1,3} s_{0,1} + 2 c_{10}) p_{0,2} p_{2,3} + (-q_{0,1} s_{1,2} \\ & - q_{1,2} s_{0,1} + 2 c_{14}) p_{0,3} p_{2,3} + (-q_{0,1} s_{0,3} - q_{0,3} s_{0,1} + 2 c_{17}) p_{1,2} p_{2,3} \\ & + (q_{0,1} s_{0,2} + q_{0,2} s_{0,1} + 2 c_{19}) p_{1,3} p_{2,3} + (q_{0,3} s_{1,3} + q_{1,3} s_{0,3} \\ & + 2 c_8) p_{0,2} p_{1,2} + (q_{0,2} s_{2,3} + q_{2,3} s_{0,2} + 2 c_4) p_{0,1} p_{1,3} + (q_{0,2} s_{1,2} \\ & + q_{1,2} s_{0,2} + 2 c_{13}) p_{0,3} p_{1,3} + (q_{0,2} s_{0,3} + q_{0,3} s_{0,2} + 2 c_{16}) p_{1,2} p_{1,3} \\ & + (q_{1,3} s_{2,3} + q_{2,3} s_{1,3} + 2 c_1) p_{0,1} p_{0,2} + (-q_{1,2} s_{2,3} - q_{2,3} s_{1,2} \end{aligned} \quad (3.4)$$

```

+ 2 c2) p0,1 p0,3 + (q1,2 s1,3 + q1,3 s1,2 + 2 c7) p0,2 p0,3 + (-q0,3 s2,3
- q2,3 s0,3 + 2 c3) p0,1 p1,2 + (-q0,1 s0,1 + c20) p2,32 + (-q1,3 s1,3 + c6) p0,22
+ (-q0,3 s0,3 + c15) p1,22 + (-q0,2 s0,2 + c18) p1,32 + (-q1,2 s1,2 + c11) p0,32
> eqsChowLines := [coeffs(%, pluckerVars), subs(p=q, pluckerRel),
subs(p=s, pluckerRel)];
eqsChowLines := [-q2,3 s2,3 + c0, -q0,1 s2,3 + q0,3 s1,2 + q1,2 s0,3 - q2,3 s0,1
+ 2 c5, -q0,2 s1,3 - q0,3 s1,2 - q1,2 s0,3 - q1,3 s0,2 + 2 c9, q0,1 s1,3 + q1,3 s0,1
+ 2 c10, -q0,1 s1,2 - q1,2 s0,1 + 2 c14, -q0,1 s0,3 - q0,3 s0,1 + 2 c17, q0,1 s0,2
+ q0,2 s0,1 + 2 c19, q0,3 s1,3 + q1,3 s0,3 + 2 c8, q0,2 s2,3 + q2,3 s0,2 + 2 c4,
q0,2 s1,2 + q1,2 s0,2 + 2 c13, q0,2 s0,3 + q0,3 s0,2 + 2 c16, q1,3 s2,3 + q2,3 s1,3
+ 2 c1, -q1,2 s2,3 - q2,3 s1,2 + 2 c2, q1,2 s1,3 + q1,3 s1,2 + 2 c7, -q0,3 s2,3
- q2,3 s0,3 + 2 c3, -q0,1 s0,1 + c20, -q1,3 s1,3 + c6, -q0,3 s0,3 + c15, -q0,2 s0,2
+ c18, -q1,2 s1,2 + c11, q0,1 q2,3 - q0,2 q1,3 + q0,3 q1,2, s0,1 s2,3 - s0,2 s1,3
+ s0,3 s1,2]

```

Again, the equations do not depend on c[12].

```

> nops(select(has, eqsChowLines, c[12]));
0

```

It only remains to eliminate the q's and the s's.

```

> [op(indets(eqsChowLines) minus {op(cvars)})], cvars;
[q0,1, q0,2, q0,3, q1,2, q1,3, q2,3, s0,1, s0,2, s0,3, s1,2, s1,3, s2,3], [c0, c1, c2, c3, c4, c5, c6,
c7, c8, c9, c10, c11, c12, c13, c14, c15, c16, c17, c18, c19, c20]

```

WARNING : the computation below requires ~10GB of RAM.

```

> # gbChowLines := fgb_gbasis_elim(eqsChowLines, 0, %, {"verb" = 3,
"index" = 10^8});

```

[Section 4] Ideal of the Chow forms of plane conics

Let us consider a plane conic in P³, given by a linear form f₁ and a quadratic form f₂.

```

> pX := stiefelMat.Vector([1,t]);

```

$$pX := \begin{bmatrix} t a_{21} + a_{11} \\ t a_{22} + a_{12} \\ t a_{23} + a_{13} \\ t a_{24} + a_{14} \end{bmatrix} \quad (4.1)$$

```

> f1 := collect(add(u[i]*pX[i], i=1..4), t, normal);

```

$$f1 := (a_{21} u_1 + a_{22} u_2 + a_{23} u_3 + a_{24} u_4) t + u_1 a_{11} + u_2 a_{12} + u_3 a_{13} + u_4 a_{14} \quad (4.2)$$

```
> f2 := collect(Transpose(pX).Matrix(4,4,(i,j) -> v[i+10*j], shape=
symmetric).pX, t, normal);
```

$$f2 := (a_{21}^2 v_{11} + 2 a_{21} a_{22} v_{21} + 2 a_{21} a_{23} v_{31} + 2 a_{21} a_{24} v_{41} + a_{22}^2 v_{22} + 2 a_{22} a_{23} v_{32} + 2 a_{22} a_{24} v_{42} + a_{23}^2 v_{33} + 2 a_{23} a_{24} v_{43} + a_{24}^2 v_{44}) t^2 + (2 a_{11} a_{21} v_{11} + 2 a_{11} a_{22} v_{21} + 2 a_{11} a_{23} v_{31} + 2 a_{11} a_{24} v_{41} + 2 a_{12} a_{21} v_{21} + 2 a_{12} a_{22} v_{22} + 2 a_{12} a_{23} v_{32} + 2 a_{12} a_{24} v_{42} + 2 a_{13} a_{21} v_{31} + 2 a_{13} a_{22} v_{32} + 2 a_{13} a_{23} v_{33} + 2 a_{13} a_{24} v_{43} + 2 a_{14} a_{21} v_{41} + 2 a_{14} a_{22} v_{42} + 2 a_{14} a_{23} v_{43} + 2 a_{14} a_{24} v_{44}) t + a_{11}^2 v_{11} + 2 a_{11} a_{12} v_{21} + 2 a_{11} a_{13} v_{31} + 2 a_{11} a_{14} v_{41} + a_{12}^2 v_{22} + 2 a_{12} a_{13} v_{32} + 2 a_{12} a_{14} v_{42} + a_{13}^2 v_{33} + 2 a_{13} a_{14} v_{43} + a_{14}^2 v_{44} \quad (4.3)$$

A line in P^3 , given by Stiefel coordinates meets the conic if and only if the following resultant vanishes:

```
> res := resultant(f1, f2, t):
```

We can express this resultant as a quadratic form in the Plucker coordinates.

```
> pluckerRes := fgb_gbasis_elim([res, seq(pluckerVars[i]-subs
(pluckerCo,pluckerVars[i]), i=1..6)], 0, stiefelVars, [op(indets
(res) minus {op(stiefelVars)}), op(pluckerVars)])[1];
```

$$\begin{aligned} pluckerRes := & p_{0,1}^2 u_1^2 v_{22} - 2 p_{0,1}^2 u_1 u_2 v_{21} + p_{0,1}^2 u_2^2 v_{11} + 2 p_{0,1} p_{0,2} u_1^2 v_{32} \\ & - 2 p_{0,1} p_{0,2} u_1 u_2 v_{31} - 2 p_{0,1} p_{0,2} u_1 u_3 v_{21} + 2 p_{0,1} p_{0,2} u_2 u_3 v_{11} + 2 p_{0,1} p_{0,3} \\ & u_1^2 v_{42} - 2 p_{0,1} p_{0,3} u_1 u_2 v_{41} - 2 p_{0,1} p_{0,3} u_1 u_4 v_{21} + 2 p_{0,1} p_{0,3} u_2 u_4 v_{11} \\ & + 2 p_{0,1} p_{1,2} u_1 u_2 v_{32} - 2 p_{0,1} p_{1,2} u_1 u_3 v_{22} - 2 p_{0,1} p_{1,2} u_2^2 v_{31} \\ & + 2 p_{0,1} p_{1,2} u_2 u_3 v_{21} + 2 p_{0,1} p_{1,3} u_1 u_2 v_{42} - 2 p_{0,1} p_{1,3} u_1 u_4 v_{22} - 2 p_{0,1} p_{1,3} \\ & u_2^2 v_{41} + 2 p_{0,1} p_{1,3} u_2 u_4 v_{21} - 2 p_{0,1} p_{2,3} u_1 u_2 v_{43} + 4 p_{0,1} p_{2,3} u_1 u_3 v_{42} \\ & - 2 p_{0,1} p_{2,3} u_1 u_4 v_{32} - 2 p_{0,1} p_{2,3} u_2 u_3 v_{41} + 4 p_{0,1} p_{2,3} u_2 u_4 v_{31} \\ & - 2 p_{0,1} p_{2,3} u_3 u_4 v_{21} + p_{0,2}^2 u_1^2 v_{33} - 2 p_{0,2}^2 u_1 u_3 v_{31} + p_{0,2}^2 u_3^2 v_{11} + 2 p_{0,2} p_{0,3} \\ & u_1^2 v_{43} - 2 p_{0,2} p_{0,3} u_1 u_3 v_{41} - 2 p_{0,2} p_{0,3} u_1 u_4 v_{31} + 2 p_{0,2} p_{0,3} u_3 u_4 v_{11} \\ & + 2 p_{0,2} p_{1,2} u_1 u_2 v_{33} - 2 p_{0,2} p_{1,2} u_1 u_3 v_{32} - 2 p_{0,2} p_{1,2} u_2 u_3 v_{31} + 2 p_{0,2} p_{1,2} \\ & u_3^2 v_{21} + 4 p_{0,2} p_{1,3} u_1 u_2 v_{43} - 2 p_{0,2} p_{1,3} u_1 u_3 v_{42} - 2 p_{0,2} p_{1,3} u_1 u_4 v_{32} \\ & - 2 p_{0,2} p_{1,3} u_2 u_3 v_{41} - 2 p_{0,2} p_{1,3} u_2 u_4 v_{31} + 4 p_{0,2} p_{1,3} u_3 u_4 v_{21} \\ & + 2 p_{0,2} p_{2,3} u_1 u_3 v_{43} - 2 p_{0,2} p_{2,3} u_1 u_4 v_{33} - 2 p_{0,2} p_{2,3} u_3^2 v_{41} \\ & + 2 p_{0,2} p_{2,3} u_3 u_4 v_{31} + p_{0,3}^2 u_1^2 v_{44} - 2 p_{0,3}^2 u_1 u_4 v_{41} + p_{0,3}^2 u_4^2 v_{11} \\ & + 2 p_{0,3} p_{1,3} u_1 u_2 v_{44} - 2 p_{0,3} p_{1,3} u_1 u_4 v_{42} - 2 p_{0,3} p_{1,3} u_2 u_4 v_{41} + 2 p_{0,3} p_{1,3} \end{aligned} \quad (4.4)$$

$$\begin{aligned}
& u_4^2 v_{21} + 2 p_{0,3} p_{2,3} u_1 u_3 v_{44} - 2 p_{0,3} p_{2,3} u_1 u_4 v_{43} - 2 p_{0,3} p_{2,3} u_3 u_4 v_{41} \\
& + 2 p_{0,3} p_{2,3} u_4^2 v_{31} + p_{1,2}^2 u_2^2 v_{33} - 2 p_{1,2}^2 u_2 u_3 v_{32} + p_{1,2}^2 u_3^2 v_{22} + 2 p_{1,2} p_{1,3} \\
& u_2^2 v_{43} - 2 p_{1,2} p_{1,3} u_2 u_3 v_{42} - 2 p_{1,2} p_{1,3} u_2 u_4 v_{32} + 2 p_{1,2} p_{1,3} u_3 u_4 v_{22} \\
& + 2 p_{1,2} p_{2,3} u_2 u_3 v_{43} - 2 p_{1,2} p_{2,3} u_2 u_4 v_{33} - 2 p_{1,2} p_{2,3} u_3^2 v_{42} \\
& + 2 p_{1,2} p_{2,3} u_3 u_4 v_{32} + p_{1,3}^2 u_2^2 v_{44} - 2 p_{1,3}^2 u_2 u_4 v_{42} + p_{1,3}^2 u_4^2 v_{22} \\
& + 2 p_{1,3} p_{2,3} u_2 u_3 v_{44} - 2 p_{1,3} p_{2,3} u_2 u_4 v_{43} - 2 p_{1,3} p_{2,3} u_3 u_4 v_{42} + 2 p_{1,3} p_{2,3} \\
& u_4^2 v_{32} + p_{2,3}^2 u_3^2 v_{44} - 2 p_{2,3}^2 u_3 u_4 v_{43} + p_{2,3}^2 u_4^2 v_{33}
\end{aligned}$$

Therefore, the quadric gQ is the Chow form of a plane conic if it equals the form above, for some f1 and f2, modulo the Plucker relation.

> Groebner[NormalForm](gQ - pluckerRes, [pluckerRel], tdeg(op
(pluckerVars)));

$$\begin{aligned}
& (-2 u_2 u_3 v_{43} + 2 u_2 u_4 v_{33} + 2 u_3^2 v_{42} - 2 u_3 u_4 v_{32} + 2 c_{17}) p_{1,2} p_{2,3} + (-2 u_2 u_3 v_{44} \\
& + 2 u_2 u_4 v_{43} + 2 u_3 u_4 v_{42} - 2 u_4^2 v_{32} + 2 c_{19}) p_{1,3} p_{2,3} + (-2 u_1^2 v_{43} + 2 u_1 u_3 v_{41} \\
& + 2 u_1 u_4 v_{31} - 2 u_3 u_4 v_{11} + 2 c_7) p_{0,2} p_{0,3} + (-2 u_1 u_2 v_{32} + 2 u_1 u_3 v_{22} + 2 \\
& u_2^2 v_{31} - 2 u_2 u_3 v_{21} + 2 c_3) p_{0,1} p_{1,2} + (-2 u_1 u_2 v_{33} + 2 u_1 u_3 v_{32} + 2 u_2 u_3 v_{31} \\
& - 2 u_3^2 v_{21} + 2 c_8) p_{0,2} p_{1,2} + (-2 u_1 u_2 v_{42} + 2 u_1 u_4 v_{22} + 2 u_2^2 v_{41} - 2 u_2 u_4 v_{21} \\
& + 2 c_4) p_{0,1} p_{1,3} + (-2 u_1 u_2 v_{44} + 2 u_1 u_4 v_{42} + 2 u_2 u_4 v_{41} - 2 u_4^2 v_{21} \\
& + 2 c_{13}) p_{0,3} p_{1,3} + (-2 u_2^2 v_{43} + 2 u_2 u_3 v_{42} + 2 u_2 u_4 v_{32} - 2 u_3 u_4 v_{22} \\
& + 2 c_{16}) p_{1,2} p_{1,3} + (-2 u_1 u_3 v_{43} + 2 u_1 u_4 v_{33} + 2 u_3^2 v_{41} - 2 u_3 u_4 v_{31} \\
& + 2 c_{10}) p_{0,2} p_{2,3} + (-2 u_1 u_3 v_{44} + 2 u_1 u_4 v_{43} + 2 u_3 u_4 v_{41} - 2 u_4^2 v_{31} \\
& + 2 c_{14}) p_{0,3} p_{2,3} + (-u_2^2 v_{44} + 2 u_2 u_4 v_{42} - u_4^2 v_{22} + c_{18}) p_{1,3}^2 + (-u_1^2 v_{33} \\
& + 2 u_1 u_3 v_{31} - u_3^2 v_{11} + c_6) p_{0,2}^2 + (-u_1^2 v_{44} + 2 u_1 u_4 v_{41} - u_4^2 v_{11} + c_{11}) p_{0,3}^2 + (- \\
& u_2^2 v_{33} + 2 u_2 u_3 v_{32} - u_3^2 v_{22} + c_{15}) p_{1,2}^2 + (-u_1^2 v_{22} + 2 u_1 u_2 v_{21} - u_2^2 v_{11} + c_0) \\
& p_{0,1}^2 + (2 u_1 u_2 v_{43} - 4 u_1 u_3 v_{42} + 2 u_1 u_4 v_{32} + 2 u_2 u_3 v_{41} - 4 u_2 u_4 v_{31} \\
& + 2 u_3 u_4 v_{21} + 2 c_5) p_{0,1} p_{2,3} + (-4 u_1 u_2 v_{43} + 2 u_1 u_3 v_{42} + 2 u_1 u_4 v_{32} \\
& + 2 u_2 u_3 v_{41} + 2 u_2 u_4 v_{31} - 4 u_3 u_4 v_{21} + 2 c_9) p_{0,2} p_{1,3} + (-2 u_1^2 v_{32} \\
& + 2 u_1 u_2 v_{31} + 2 u_1 u_3 v_{21} - 2 u_2 u_3 v_{11} + 2 c_1) p_{0,1} p_{0,2} + (-2 u_1^2 v_{42} \\
& + 2 u_1 u_2 v_{41} + 2 u_1 u_4 v_{21} - 2 u_2 u_4 v_{11} + 2 c_2) p_{0,1} p_{0,3} + (-u_3^2 v_{44} + 2 u_3 u_4 v_{43} \\
& - u_4^2 v_{33} + c_{20}) p_{2,3}^2
\end{aligned} \tag{4.5}$$

[Section 5] Ideal of the Hurwitz forms of quadrics in P^3

A line in P^3 , given by Stiefel coordinates is tangent to the quadric defined by f_2 if and only if the following discriminant vanishes:

$$\begin{aligned} & \text{disc} := \text{collect}(\text{discrim}(f_2, t), \text{stiefelVars}, \text{'distributed'}); \\ \text{disc} := & \left(-4 v_{11} v_{22} + 4 v_{21}^2 \right) a_{11}^2 a_{22}^2 + \left(-4 v_{11} v_{33} + 4 v_{31}^2 \right) a_{11}^2 a_{23}^2 + \left(-4 v_{11} v_{44} + 4 v_{41}^2 \right) a_{11}^2 a_{24}^2 + \left(-4 v_{11} v_{22} + 4 v_{21}^2 \right) a_{12}^2 a_{21}^2 + \left(-4 v_{22} v_{33} + 4 v_{32}^2 \right) a_{12}^2 a_{23}^2 + \left(-4 v_{22} v_{44} + 4 v_{42}^2 \right) a_{12}^2 a_{24}^2 + \left(-4 v_{11} v_{33} + 4 v_{31}^2 \right) a_{13}^2 a_{21}^2 + \left(-4 v_{22} v_{33} + 4 v_{32}^2 \right) a_{13}^2 a_{23}^2 + \left(-4 v_{33} v_{44} + 4 v_{43}^2 \right) a_{13}^2 a_{24}^2 + \left(-4 v_{11} v_{44} + 4 v_{41}^2 \right) a_{14}^2 a_{21}^2 + \left(-4 v_{22} v_{44} + 4 v_{42}^2 \right) a_{14}^2 a_{22}^2 + \left(-4 v_{33} v_{44} + 4 v_{43}^2 \right) a_{14}^2 a_{23}^2 + \left(8 v_{22} v_{43} \right. \\ & - 8 v_{32} v_{42} \left. \right) a_{12} a_{14} a_{22} a_{23} + \left(8 v_{22} v_{44} - 8 v_{42}^2 \right) a_{12} a_{14} a_{22} a_{24} + \left(8 v_{32} v_{44} \right. \\ & - 8 v_{42} v_{43} \left. \right) a_{12} a_{14} a_{23} a_{24} + \left(-16 v_{21} v_{43} + 8 v_{31} v_{42} + 8 v_{32} v_{41} \right) a_{13} a_{14} a_{21} a_{22} \\ & + \left(-8 v_{31} v_{43} + 8 v_{33} v_{41} \right) a_{13} a_{14} a_{21} a_{23} + \left(8 v_{31} v_{44} - 8 v_{41} v_{43} \right) a_{13} a_{14} a_{21} a_{24} \\ & + \left(-8 v_{32} v_{43} + 8 v_{33} v_{42} \right) a_{13} a_{14} a_{22} a_{23} + \left(8 v_{32} v_{44} - 8 v_{42} v_{43} \right) a_{13} a_{14} a_{22} a_{24} \\ & + \left(8 v_{33} v_{44} - 8 v_{43}^2 \right) a_{13} a_{14} a_{23} a_{24} + \left(8 v_{11} v_{22} - 8 v_{21}^2 \right) a_{11} a_{12} a_{21} a_{22} \\ & + \left(8 v_{11} v_{32} - 8 v_{21} v_{31} \right) a_{11} a_{12} a_{21} a_{23} + \left(8 v_{11} v_{42} - 8 v_{21} v_{41} \right) a_{11} a_{12} a_{21} a_{24} \\ & + \left(-8 v_{21} v_{32} + 8 v_{22} v_{31} \right) a_{11} a_{12} a_{22} a_{23} + \left(-8 v_{21} v_{42} + 8 v_{22} v_{41} \right) a_{11} a_{12} a_{22} a_{24} \\ & + \left(-16 v_{21} v_{43} + 8 v_{31} v_{42} + 8 v_{32} v_{41} \right) a_{11} a_{12} a_{23} a_{24} + \left(8 v_{11} v_{32} \right. \\ & - 8 v_{21} v_{31} \left. \right) a_{11} a_{13} a_{21} a_{22} + \left(8 v_{11} v_{33} - 8 v_{31}^2 \right) a_{11} a_{13} a_{21} a_{23} + \left(8 v_{11} v_{43} \right. \\ & - 8 v_{31} v_{41} \left. \right) a_{11} a_{13} a_{21} a_{24} + \left(8 v_{21} v_{33} - 8 v_{31} v_{32} \right) a_{11} a_{13} a_{22} a_{23} + \left(8 v_{21} v_{43} \right. \\ & - 16 v_{31} v_{42} + 8 v_{32} v_{41} \left. \right) a_{11} a_{13} a_{22} a_{24} + \left(-8 v_{31} v_{43} + 8 v_{33} v_{41} \right) a_{11} a_{13} a_{23} a_{24} \\ & + \left(8 v_{11} v_{42} - 8 v_{21} v_{41} \right) a_{11} a_{14} a_{21} a_{22} + \left(8 v_{11} v_{43} - 8 v_{31} v_{41} \right) a_{11} a_{14} a_{21} a_{23} \\ & + \left(8 v_{11} v_{44} - 8 v_{41}^2 \right) a_{11} a_{14} a_{21} a_{24} + \left(8 v_{21} v_{43} + 8 v_{31} v_{42} \right. \\ & - 16 v_{32} v_{41} \left. \right) a_{11} a_{14} a_{22} a_{23} + \left(8 v_{21} v_{44} - 8 v_{41} v_{42} \right) a_{11} a_{14} a_{22} a_{24} + \left(8 v_{31} v_{44} \right. \\ & - 8 v_{41} v_{43} \left. \right) a_{11} a_{14} a_{23} a_{24} + \left(-8 v_{21} v_{32} + 8 v_{22} v_{31} \right) a_{12} a_{13} a_{21} a_{22} + \left(8 v_{21} v_{33} \right. \\ & - 8 v_{31} v_{32} \left. \right) a_{12} a_{13} a_{21} a_{23} + \left(8 v_{21} v_{43} + 8 v_{31} v_{42} - 16 v_{32} v_{41} \right) a_{12} a_{13} a_{21} a_{24} \\ & + \left(8 v_{22} v_{33} - 8 v_{32}^2 \right) a_{12} a_{13} a_{22} a_{23} + \left(8 v_{22} v_{43} - 8 v_{32} v_{42} \right) a_{12} a_{13} a_{22} a_{24} + \left(-8 v_{32} v_{43} + 8 v_{33} v_{42} \right) a_{12} a_{13} a_{23} a_{24} + \left(-8 v_{21} v_{42} + 8 v_{22} v_{41} \right) a_{12} a_{14} a_{21} a_{22} \\ & + \left(8 v_{21} v_{43} - 16 v_{31} v_{42} + 8 v_{32} v_{41} \right) a_{12} a_{14} a_{21} a_{23} + \left(8 v_{21} v_{44} \right. \\ & - 8 v_{41} v_{42} \left. \right) a_{12} a_{14} a_{21} a_{24} + \left(-8 v_{11} v_{32} + 8 v_{21} v_{31} \right) a_{11}^2 a_{22} a_{23} + \left(-8 v_{11} v_{42} \right. \\ & + 8 v_{21} v_{41} \left. \right) a_{11}^2 a_{22} a_{24} + \left(-8 v_{11} v_{43} + 8 v_{31} v_{41} \right) a_{11}^2 a_{23} a_{24} + \left(-8 v_{21} v_{33} \right. \end{aligned} \tag{5.1}$$

$$\begin{aligned}
& + 8 v_{31} v_{32}) a_{11} a_{12} a_{23}^2 + (-8 v_{21} v_{44} + 8 v_{41} v_{42}) a_{11} a_{12} a_{24}^2 + (8 v_{21} v_{32} \\
& - 8 v_{22} v_{31}) a_{11} a_{13} a_{22}^2 + (-8 v_{31} v_{44} + 8 v_{41} v_{43}) a_{11} a_{13} a_{24}^2 + (8 v_{21} v_{42} \\
& - 8 v_{22} v_{41}) a_{11} a_{14} a_{22}^2 + (8 v_{31} v_{43} - 8 v_{33} v_{41}) a_{11} a_{14} a_{23}^2 + (8 v_{21} v_{32} \\
& - 8 v_{22} v_{31}) a_{12}^2 a_{21} a_{23} + (8 v_{21} v_{42} - 8 v_{22} v_{41}) a_{12}^2 a_{21} a_{24} + (-8 v_{22} v_{43} \\
& + 8 v_{32} v_{42}) a_{12}^2 a_{23} a_{24} + (-8 v_{11} v_{32} + 8 v_{21} v_{31}) a_{12} a_{13} a_{21}^2 + (-8 v_{32} v_{44} \\
& + 8 v_{42} v_{43}) a_{12} a_{13} a_{24}^2 + (-8 v_{11} v_{42} + 8 v_{21} v_{41}) a_{12} a_{14} a_{21}^2 + (8 v_{32} v_{43} \\
& - 8 v_{33} v_{42}) a_{12} a_{14} a_{23}^2 + (-8 v_{21} v_{33} + 8 v_{31} v_{32}) a_{13}^2 a_{21} a_{22} + (8 v_{31} v_{43} \\
& - 8 v_{33} v_{41}) a_{13}^2 a_{21} a_{24} + (8 v_{32} v_{43} - 8 v_{33} v_{42}) a_{13}^2 a_{22} a_{24} + (-8 v_{11} v_{43} \\
& + 8 v_{31} v_{41}) a_{13} a_{14} a_{21}^2 + (-8 v_{22} v_{43} + 8 v_{32} v_{42}) a_{13} a_{14} a_{22}^2 + (-8 v_{21} v_{44} \\
& + 8 v_{41} v_{42}) a_{14}^2 a_{21} a_{22} + (-8 v_{31} v_{44} + 8 v_{41} v_{43}) a_{14}^2 a_{21} a_{23} + (-8 v_{32} v_{44} \\
& + 8 v_{42} v_{43}) a_{14}^2 a_{22} a_{23}
\end{aligned}$$

This can be expressed in terms of the Plucker coordinates.

```

> pluckerDisc := fgb_gbasis_elim([disc, seq(pluckerVars[i]-subs
(pluckerCo,pluckerVars[i]), i=1..6)], 0, stiefelVars, [op(indets
(disc) minus {op(stiefelVars)}), op(pluckerVars)])[1];

```

$$\begin{aligned}
pluckerDisc := & -p_{0,1}^2 v_{11} v_{22} + p_{0,1}^2 v_{21}^2 - 2 p_{0,1} p_{0,2} v_{11} v_{32} + 2 p_{0,1} p_{0,2} v_{21} v_{31} \\
& - 2 p_{0,1} p_{0,3} v_{11} v_{42} + 2 p_{0,1} p_{0,3} v_{21} v_{41} - 2 p_{0,1} p_{1,2} v_{21} v_{32} + 2 p_{0,1} p_{1,2} v_{22} v_{31} \\
& - 2 p_{0,1} p_{1,3} v_{21} v_{42} + 2 p_{0,1} p_{1,3} v_{22} v_{41} + 2 p_{0,1} p_{2,3} v_{21} v_{43} - 4 p_{0,1} p_{2,3} v_{31} v_{42} \\
& + 2 p_{0,1} p_{2,3} v_{32} v_{41} - p_{0,2}^2 v_{11} v_{33} + p_{0,2}^2 v_{31}^2 - 2 p_{0,2} p_{0,3} v_{11} v_{43} \\
& + 2 p_{0,2} p_{0,3} v_{31} v_{41} - 2 p_{0,2} p_{1,2} v_{21} v_{33} + 2 p_{0,2} p_{1,2} v_{31} v_{32} - 4 p_{0,2} p_{1,3} v_{21} v_{43} \\
& + 2 p_{0,2} p_{1,3} v_{31} v_{42} + 2 p_{0,2} p_{1,3} v_{32} v_{41} - 2 p_{0,2} p_{2,3} v_{31} v_{43} + 2 p_{0,2} p_{2,3} v_{33} v_{41} \\
& - p_{0,3}^2 v_{11} v_{44} + p_{0,3}^2 v_{41}^2 - 2 p_{0,3} p_{1,3} v_{21} v_{44} + 2 p_{0,3} p_{1,3} v_{41} v_{42} \\
& - 2 p_{0,3} p_{2,3} v_{31} v_{44} + 2 p_{0,3} p_{2,3} v_{41} v_{43} - p_{1,2}^2 v_{22} v_{33} + p_{1,2}^2 v_{32}^2 \\
& - 2 p_{1,2} p_{1,3} v_{22} v_{43} + 2 p_{1,2} p_{1,3} v_{32} v_{42} - 2 p_{1,2} p_{2,3} v_{32} v_{43} + 2 p_{1,2} p_{2,3} v_{33} v_{42} \\
& - p_{1,3}^2 v_{22} v_{44} + p_{1,3}^2 v_{42}^2 - 2 p_{1,3} p_{2,3} v_{32} v_{44} + 2 p_{1,3} p_{2,3} v_{42} v_{43} - p_{2,3}^2 v_{33} v_{44} \\
& + p_{2,3}^2 v_{43}^2
\end{aligned} \tag{5.2}$$

As above, we look for the conditions that gQ equals pluckerDisc modulo the Plucker relation.

```

> Groebner[NormalForm](gQ - pluckerDisc, [pluckerRel], tdeg(op
(pluckerVars)));

```

$$\begin{aligned}
& (2 v_{21} v_{42} - 2 v_{22} v_{41} + 2 c_4) p_{0,1} p_{1,3} + (2 v_{21} v_{44} - 2 v_{41} v_{42} + 2 c_{13}) p_{0,3} p_{1,3} \\
& + (2 v_{22} v_{43} - 2 v_{32} v_{42} + 2 c_{16}) p_{1,2} p_{1,3} + (-2 v_{21} v_{43} + 4 v_{31} v_{42} - 2 v_{32} v_{41}
\end{aligned} \tag{5.3}$$

$$\begin{aligned}
& + 2c_5) p_{0,1} p_{2,3} + (v_{22} v_{44} - v_{42}^2 + c_{18}) p_{1,3}^2 + (4v_{21} v_{43} - 2v_{31} v_{42} - 2v_{32} v_{41} \\
& + 2c_9) p_{0,2} p_{1,3} + (v_{11} v_{22} - v_{21}^2 + c_0) p_{0,1}^2 + (v_{11} v_{33} - v_{31}^2 + c_6) p_{0,2}^2 \\
& + (v_{11} v_{44} - v_{41}^2 + c_{11}) p_{0,3}^2 + (v_{22} v_{33} - v_{32}^2 + c_{15}) p_{1,2}^2 + (2v_{11} v_{32} - 2v_{21} v_{31} \\
& + 2c_1) p_{0,1} p_{0,2} + (2v_{11} v_{42} - 2v_{21} v_{41} + 2c_2) p_{0,1} p_{0,3} + (2v_{11} v_{43} - 2v_{31} v_{41} \\
& + 2c_7) p_{0,2} p_{0,3} + (2v_{21} v_{32} - 2v_{22} v_{31} + 2c_3) p_{0,1} p_{1,2} + (2v_{21} v_{33} - 2v_{31} v_{32} \\
& + 2c_8) p_{0,2} p_{1,2} + (v_{33} v_{44} - v_{43}^2 + c_{20}) p_{2,3}^2 + (2v_{31} v_{43} - 2v_{33} v_{41} \\
& + 2c_{10}) p_{0,2} p_{2,3} + (2v_{31} v_{44} - 2v_{41} v_{43} + 2c_{14}) p_{0,3} p_{2,3} + (2v_{32} v_{43} \\
& - 2v_{33} v_{42} + 2c_{17}) p_{1,2} p_{2,3} + (2v_{32} v_{44} - 2v_{42} v_{43} + 2c_{19}) p_{1,3} p_{2,3}
\end{aligned}$$

```
> eqsHurwitz := [coeffs(%, pluckerVars)];
```

$$\begin{aligned}
eqsHurwitz := & \left[2v_{21} v_{42} - 2v_{22} v_{41} + 2c_4, 2v_{21} v_{44} - 2v_{41} v_{42} + 2c_{13}, 2v_{22} v_{43} \right. \\
& - 2v_{32} v_{42} + 2c_{16}, -2v_{21} v_{43} + 4v_{31} v_{42} - 2v_{32} v_{41} + 2c_5, v_{22} v_{44} - v_{42}^2 + c_{18}, \\
& 4v_{21} v_{43} - 2v_{31} v_{42} - 2v_{32} v_{41} + 2c_9, v_{11} v_{22} - v_{21}^2 + c_0, v_{11} v_{33} - v_{31}^2 + c_6, v_{11} v_{44} \\
& - v_{41}^2 + c_{11}, v_{22} v_{33} - v_{32}^2 + c_{15}, 2v_{11} v_{32} - 2v_{21} v_{31} + 2c_1, 2v_{11} v_{42} - 2v_{21} v_{41} \\
& + 2c_2, 2v_{11} v_{43} - 2v_{31} v_{41} + 2c_7, 2v_{21} v_{32} - 2v_{22} v_{31} + 2c_3, 2v_{21} v_{33} - 2v_{31} v_{32} \\
& + 2c_8, v_{33} v_{44} - v_{43}^2 + c_{20}, 2v_{31} v_{43} - 2v_{33} v_{41} + 2c_{10}, 2v_{31} v_{44} - 2v_{41} v_{43} \\
& \left. + 2c_{14}, 2v_{32} v_{43} - 2v_{33} v_{42} + 2c_{17}, 2v_{32} v_{44} - 2v_{42} v_{43} + 2c_{19} \right]
\end{aligned} \tag{5.4}$$

WARNING : the computation below requires some RAM.

```
> # gbHurwitz := fgb_gbasis_elim(eqsHurwitz, 0, [op(indets
(eqsHurwitz) minus {op(cvars)})], cvars0, {"verb" = 3, "index" =
10^8}):
```

▼ [Section 5bis] Ideal of the Hurwitz forms of quadrics in P^3 (alternative method)

See B. Sturmfels: The Hurwitz form of a projective variety, Example 2.2, for more details about the method.

```
> sym4 := Matrix(4, (i,j) -> d[i+10*j], shape = symmetric);
```

$$sym4 := \begin{bmatrix} d_{11} & d_{21} & d_{31} & d_{41} \\ d_{21} & d_{22} & d_{32} & d_{42} \\ d_{31} & d_{32} & d_{33} & d_{43} \\ d_{41} & d_{42} & d_{43} & d_{44} \end{bmatrix} \tag{6.1}$$

```
> choose24 := combinat[choose]([1,2,3,4], 2);
```

(6.2)

$$\text{choose24} := [[1, 2], [1, 3], [1, 4], [2, 3], [2, 4], [3, 4]] \quad (6.2)$$

> **extSym4** := **Matrix**(6, (i,j) -> **Determinant**(**sym4**[**choose24**[i], **choose24**[j]]));

$$\text{extSym4} := \left[\left[d_{11} d_{22} - d_{21}^2, d_{11} d_{32} - d_{21} d_{31}, d_{11} d_{42} - d_{21} d_{41}, d_{21} d_{32} - d_{22} d_{31}, \right. \right. \quad (6.3)$$

$$d_{21} d_{42} - d_{22} d_{41}, d_{31} d_{42} - d_{32} d_{41} \Big],$$

$$\left[d_{11} d_{32} - d_{21} d_{31}, d_{11} d_{33} - d_{31}^2, d_{11} d_{43} - d_{31} d_{41}, d_{21} d_{33} - d_{31} d_{32}, d_{21} d_{43} \right. \\ \left. - d_{32} d_{41}, d_{31} d_{43} - d_{33} d_{41} \right],$$

$$\left[d_{11} d_{42} - d_{21} d_{41}, d_{11} d_{43} - d_{31} d_{41}, d_{11} d_{44} - d_{41}^2, d_{21} d_{43} - d_{31} d_{42}, d_{21} d_{44} \right. \\ \left. - d_{41} d_{42}, d_{31} d_{44} - d_{41} d_{43} \right],$$

$$\left[d_{21} d_{32} - d_{22} d_{31}, d_{21} d_{33} - d_{31} d_{32}, d_{21} d_{43} - d_{31} d_{42}, d_{22} d_{33} - d_{32}^2, d_{22} d_{43} \right. \\ \left. - d_{32} d_{42}, d_{32} d_{43} - d_{33} d_{42} \right],$$

$$\left[d_{21} d_{42} - d_{22} d_{41}, d_{21} d_{43} - d_{32} d_{41}, d_{21} d_{44} - d_{41} d_{42}, d_{22} d_{43} - d_{32} d_{42}, d_{22} d_{44} - \right. \\ \left. d_{42}^2, d_{32} d_{44} - d_{42} d_{43} \right],$$

$$\left[d_{31} d_{42} - d_{32} d_{41}, d_{31} d_{43} - d_{33} d_{41}, d_{31} d_{44} - d_{41} d_{43}, d_{32} d_{43} - d_{33} d_{42}, d_{32} d_{44} \right. \\ \left. - d_{42} d_{43}, d_{33} d_{44} - d_{43}^2 \right]]$$

> **gHurwitz** := **collect**(**Vector**[**row**](**pluckerVars**).**extSym4**.**Vector**(**pluckerVars**), **pluckerVars**, **distributed**);

$$g\text{Hurwitz} := \left(d_{11} d_{22} - d_{21}^2 \right) p_{0,1}^2 + \left(2 d_{21} d_{42} - 2 d_{22} d_{41} \right) p_{1,3} p_{0,1} + \left(2 d_{31} d_{42} \right. \quad (6.4)$$

$$- 2 d_{32} d_{41} \Big) p_{2,3} p_{0,1} + \left(2 d_{11} d_{32} - 2 d_{21} d_{31} \right) p_{0,2} p_{0,1} + \left(2 d_{11} d_{42} \right.$$

$$- 2 d_{21} d_{41} \Big) p_{0,3} p_{0,1} + \left(2 d_{21} d_{32} - 2 d_{22} d_{31} \right) p_{1,2} p_{0,1} + \left(d_{11} d_{33} - d_{31}^2 \right) p_{0,2}^2$$

$$+ \left(2 d_{21} d_{43} - 2 d_{32} d_{41} \right) p_{1,3} p_{0,2} + \left(2 d_{31} d_{43} - 2 d_{33} d_{41} \right) p_{2,3} p_{0,2}$$

$$+ \left(2 d_{11} d_{43} - 2 d_{31} d_{41} \right) p_{0,3} p_{0,2} + \left(2 d_{21} d_{33} - 2 d_{31} d_{32} \right) p_{1,2} p_{0,2} + \left(d_{11} d_{44} \right.$$

$$- d_{41}^2 \Big) p_{0,3}^2 + \left(2 d_{21} d_{44} - 2 d_{41} d_{42} \right) p_{1,3} p_{0,3} + \left(2 d_{31} d_{44} - 2 d_{41} d_{43} \right) p_{2,3} p_{0,3}$$

$$+ \left(2 d_{21} d_{43} - 2 d_{31} d_{42} \right) p_{1,2} p_{0,3} + \left(d_{22} d_{33} - d_{32}^2 \right) p_{1,2}^2 + \left(2 d_{22} d_{43} \right.$$

$$- 2 d_{32} d_{42} \Big) p_{1,3} p_{1,2} + \left(2 d_{32} d_{43} - 2 d_{33} d_{42} \right) p_{2,3} p_{1,2} + \left(d_{22} d_{44} - d_{42}^2 \right) p_{1,3}^2$$

$$+ \left(2 d_{32} d_{44} - 2 d_{42} d_{43} \right) p_{2,3} p_{1,3} + p_{2,3}^2 \left(d_{33} d_{44} - d_{43}^2 \right)$$

> **Groebner**[**Reduce**](**gQ** - **gHurwitz**, [**pluckerRel**], **tdeg**(**op**(**pluckerVars**)));

$$\left(-d_{11} d_{33} + d_{31}^2 + c_6 \right) p_{0,2}^2 + \left(-d_{11} d_{44} + d_{41}^2 + c_{11} \right) p_{0,3}^2 + \left(-d_{22} d_{33} + d_{32}^2 + c_{15} \right) \quad (6.5)$$

$$p_{1,2}^2 + \left(-d_{22} d_{44} + d_{42}^2 + c_{18} \right) p_{1,3}^2 + \left(-d_{11} d_{22} + d_{21}^2 + c_0 \right) p_{0,1}^2 + \left(-2 d_{11} d_{43} \right.$$

$$+ 2 d_{31} d_{41} + 2 c_7 \Big) p_{0,3} p_{0,2} + \left(-2 d_{32} d_{43} + 2 d_{33} d_{42} + 2 c_{17} \right) p_{2,3} p_{1,2} + \left(\right.$$

$$\begin{aligned}
& -2 d_{21} d_{33} + 2 d_{31} d_{32} + 2 c_8) p_{1,2} p_{0,2} + (-2 d_{21} d_{42} + 2 d_{22} d_{41} + 2 c_4) p_{1,3} p_{0,1} \\
& + (-2 d_{31} d_{44} + 2 d_{41} d_{43} + 2 c_{14}) p_{2,3} p_{0,3} + (-4 d_{21} d_{43} + 2 d_{31} d_{42} + 2 d_{32} d_{41} \\
& + 2 c_9) p_{1,3} p_{0,2} + (-2 d_{32} d_{44} + 2 d_{42} d_{43} + 2 c_{19}) p_{2,3} p_{1,3} + (-2 d_{31} d_{43} \\
& + 2 d_{33} d_{41} + 2 c_{10}) p_{2,3} p_{0,2} + (-2 d_{22} d_{43} + 2 d_{32} d_{42} + 2 c_{16}) p_{1,3} p_{1,2} + (\\
& -2 d_{21} d_{32} + 2 d_{22} d_{31} + 2 c_3) p_{1,2} p_{0,1} + (-2 d_{21} d_{44} + 2 d_{41} d_{42} \\
& + 2 c_{13}) p_{1,3} p_{0,3} + (-2 d_{11} d_{42} + 2 d_{21} d_{41} + 2 c_2) p_{0,3} p_{0,1} + (-2 d_{11} d_{32} \\
& + 2 d_{21} d_{31} + 2 c_1) p_{0,2} p_{0,1} + (2 d_{21} d_{43} - 4 d_{31} d_{42} + 2 d_{32} d_{41} + 2 c_5) p_{2,3} p_{0,1} \\
& + (-d_{33} d_{44} + d_{43}^2 + c_{20}) p_{2,3}^2
\end{aligned}$$

> eqsHurwitzAlt := [coeffs(%, pluckerVars)];

$$\begin{aligned}
eqsHurwitzAlt := & \left[-d_{11} d_{33} + d_{31}^2 + c_6, -d_{11} d_{44} + d_{41}^2 + c_{11}, -d_{22} d_{33} + d_{32}^2 + c_{15}, \right. \\
& -d_{22} d_{44} + d_{42}^2 + c_{18}, -d_{11} d_{22} + d_{21}^2 + c_0, -2 d_{11} d_{43} + 2 d_{31} d_{41} + 2 c_7, -2 d_{32} d_{43} \\
& + 2 d_{33} d_{42} + 2 c_{17}, -2 d_{21} d_{33} + 2 d_{31} d_{32} + 2 c_8, -2 d_{21} d_{42} + 2 d_{22} d_{41} + 2 c_4, \\
& -2 d_{31} d_{44} + 2 d_{41} d_{43} + 2 c_{14}, -4 d_{21} d_{43} + 2 d_{31} d_{42} + 2 d_{32} d_{41} + 2 c_9, -2 d_{32} d_{44} \\
& + 2 d_{42} d_{43} + 2 c_{19}, -2 d_{31} d_{43} + 2 d_{33} d_{41} + 2 c_{10}, -2 d_{22} d_{43} + 2 d_{32} d_{42} + 2 c_{16}, \\
& -2 d_{21} d_{32} + 2 d_{22} d_{31} + 2 c_3, -2 d_{21} d_{44} + 2 d_{41} d_{42} + 2 c_{13}, -2 d_{11} d_{42} + 2 d_{21} d_{41} \\
& + 2 c_2, -2 d_{11} d_{32} + 2 d_{21} d_{31} + 2 c_1, 2 d_{21} d_{43} - 4 d_{31} d_{42} + 2 d_{32} d_{41} + 2 c_5, \\
& \left. -d_{33} d_{44} + d_{43}^2 + c_{20} \right]
\end{aligned} \tag{6.6}$$

> nops(select(has, eqsHurwitzAlt, c[12]));

0

(6.7)

It only remains to eliminate the d's

WARNING : the computation below requires some RAM.

> # gbHurwitzAlt := fgb_gbasis_elim(eqsHurwitzAlt, 0, [op(indets
(eqsHurwitzAlt) minus {op(cvars)}]), cvars0, {"verb" = 3, "index"
= 10^8});

> # evalb(gbHurwitz = gbHurwitzAlt);

Section 6] Ideal of Chow forms that are squares

> Groebner[Reduce](gQ - add(e[i]*pluckerVars[i], i=1..6)^2,
[pluckerRel], tdeg(op(pluckerVars)));

$$\begin{aligned}
& (-e_2^2 + c_6) p_{0,2}^2 + (-e_3^2 + c_{11}) p_{0,3}^2 + (-e_4^2 + c_{15}) p_{1,2}^2 + (-e_5^2 + c_{18}) p_{1,3}^2 + (-e_1^2 \\
& + c_0) p_{0,1}^2 + (-2 e_2 e_3 + 2 c_7) p_{0,3} p_{0,2} + (-2 e_4 e_6 + 2 c_{17}) p_{2,3} p_{1,2} + (\\
& -2 e_2 e_4 + 2 c_8) p_{1,2} p_{0,2} + (-2 e_1 e_5 + 2 c_4) p_{1,3} p_{0,1} + (-2 e_3 e_6
\end{aligned} \tag{7.1}$$

[illegible][illegible]

This is Proposition 2 in “Computing the Chow variety of quadratic space curves”

[illegible]

```
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]
```

[Section 8] Primary decomposition of the coisotropic ideal

We now check that the coisotropic ideal is the intersection of the ideals corresponding to Chow forms of conics, Chow form of skew lines and Hurwitz form of quadrics.

```
> eqsInter := [1-u[1]-u[2]-u[3], seq(u[1]*p, p in gbSquare), seq(u[2]*p, p in gbHurwitz), seq(u[3]*p, p in gbChowLines)]:
```

This computes Groebner basis of the intersection.

```
> gbInter := fgb_gbasis_elim(eqsInter, 0, [u[1], u[2], u[3]], cvars0):
```

This is Proposition 1 in “Computing the Chow variety of quadratic space curves”

```
> evalb(gbInter = gbCoiso);
```

true

(9.1)

[Section 9] Ideal of the Chow variety

The Chow variety $G(2,2,4)$ is formed by the Chow forms of all plane cubics and all pairs of skew lines.

```
> eqsChowForm := [seq((1-t)*p, p in gbChowLines), seq(t*p, p in gbChowConic)]:
```

```
> gbChowForm := fgb_gbasis_elim(eqsChowForm, 0, [t], cvars0): nops(%);
```

348

(10.1)

[Section 10] The integrability ideal

A few functions to deal with differential forms.

```
> read "Df.mpl": with(Df[MM]):
```

Here, we deal with affine the following affine chart of the Grassmannian.

```
> affineCo := [a[11]=1, a[21]=0, a[12]=0, a[22]=1];
```

$$\text{affineCo} := [a_{11} = 1, a_{21} = 0, a_{12} = 0, a_{22} = 1]$$

(11.1)

```
> affineVars := remove(type, subs(affineCo, stiefelVars), integer);
```

$$\text{affineVars} := [a_{13}, a_{14}, a_{23}, a_{24}]$$

(11.2)

```
> pluckerToAffine := subs(affineCo, pluckerCo);
```

$$\begin{aligned} \text{pluckerToAffine} := & [p_{0,1} = 1, p_{0,2} = a_{23}, p_{0,3} = a_{24}, p_{1,2} = -a_{13}, p_{1,3} = -a_{14}, p_{2,3} \\ & = a_{13} a_{24} - a_{14} a_{23}] \end{aligned}$$

(11.3)

```
> alg := map(v->d[v], convert(affineVars, set));
```

$$\text{alg} := \{d_{a_{13}}, d_{a_{14}}, d_{a_{23}}, d_{a_{24}}\}$$

(11.4)

Note that we use dual coordinates.

$$\begin{aligned}
 &> \text{affgQ} := \text{subs}(\text{pluckerDual}, \text{pluckerToAffine}, \text{gQ}); \\
 \text{affgQ} &:= c_0 (a_{13} a_{24} - a_{14} a_{23})^2 - 2 c_4 a_{23} (a_{13} a_{24} - a_{14} a_{23}) + (2 c_5 \\
 &\quad + 2 c_{12}) (a_{13} a_{24} - a_{14} a_{23}) + 2 c_1 a_{14} (a_{13} a_{24} - a_{14} a_{23}) - 2 c_2 a_{13} (a_{13} a_{24} \\
 &\quad - a_{14} a_{23}) + 2 c_3 a_{24} (a_{13} a_{24} - a_{14} a_{23}) + c_6 a_{14}^2 - (2 c_9 - 2 c_{12}) a_{14} a_{23} \\
 &\quad + 2 c_{10} a_{14} - 2 c_7 a_{13} a_{14} + 2 c_8 a_{24} a_{14} + c_{11} a_{13}^2 - 2 c_{12} a_{13} a_{24} + 2 c_{13} a_{23} a_{13} \\
 &\quad - 2 c_{14} a_{13} + c_{15} a_{24}^2 - 2 c_{16} a_{23} a_{24} + 2 c_{17} a_{24} + c_{18} a_{23}^2 - 2 c_{19} a_{23} + c_{20}
 \end{aligned} \tag{11.5}$$

$$\begin{aligned}
 &> \text{affineMat} := \text{subs}(\text{affineCo}, \text{stiefelMat}); \\
 \text{affineMat} &:= \begin{bmatrix} 1 & 0 \\ 0 & 1 \\ a_{13} & a_{23} \\ a_{14} & a_{24} \end{bmatrix}
 \end{aligned} \tag{11.6}$$

$$\begin{aligned}
 &> \text{alphas} := [\text{seq}(\text{seq}(\text{add}(\text{diff}(\text{affgQ}, \text{affineMat}[1+2, i]) * \text{d}[\text{affineMat}[1+2, j]], 1=1..2), j=1..2), i=1..2)]; \\
 \text{alphas} &:= \left[\begin{aligned}
 &(2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{24} - 2 c_4 a_{23} a_{24} + (2 c_5 + 2 c_{12}) a_{24} \\
 &\quad + 2 c_1 a_{14} a_{24} - 2 c_2 (a_{13} a_{24} - a_{14} a_{23}) - 2 c_2 a_{13} a_{24} + 2 c_3 a_{24}^2 - 2 c_7 a_{14} \\
 &\quad + 2 c_{11} a_{13} - 2 c_{12} a_{24} + 2 c_{13} a_{23} - 2 c_{14}) d_{a_{13}} + (-2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{23} \\
 &\quad + 2 c_4 a_{23}^2 - (2 c_5 + 2 c_{12}) a_{23} + 2 c_1 (a_{13} a_{24} - a_{14} a_{23}) - 2 c_1 a_{14} a_{23} \\
 &\quad + 2 c_2 a_{13} a_{23} - 2 c_3 a_{24} a_{23} + 2 c_6 a_{14} - (2 c_9 - 2 c_{12}) a_{23} + 2 c_{10} - 2 c_7 a_{13} \\
 &\quad + 2 c_8 a_{24}) d_{a_{14}}, (2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{24} - 2 c_4 a_{23} a_{24} + (2 c_5 + 2 c_{12}) a_{24} \\
 &\quad + 2 c_1 a_{14} a_{24} - 2 c_2 (a_{13} a_{24} - a_{14} a_{23}) - 2 c_2 a_{13} a_{24} + 2 c_3 a_{24}^2 - 2 c_7 a_{14} \\
 &\quad + 2 c_{11} a_{13} - 2 c_{12} a_{24} + 2 c_{13} a_{23} - 2 c_{14}) d_{a_{23}} + (-2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{23} \\
 &\quad + 2 c_4 a_{23}^2 - (2 c_5 + 2 c_{12}) a_{23} + 2 c_1 (a_{13} a_{24} - a_{14} a_{23}) - 2 c_1 a_{14} a_{23} \\
 &\quad + 2 c_2 a_{13} a_{23} - 2 c_3 a_{24} a_{23} + 2 c_6 a_{14} - (2 c_9 - 2 c_{12}) a_{23} + 2 c_{10} - 2 c_7 a_{13} \\
 &\quad + 2 c_8 a_{24}) d_{a_{24}}, (-2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{14} - 2 c_4 (a_{13} a_{24} - a_{14} a_{23}) \\
 &\quad + 2 c_4 a_{23} a_{14} - (2 c_5 + 2 c_{12}) a_{14} - 2 c_1 a_{14}^2 + 2 c_2 a_{13} a_{14} - 2 c_3 a_{24} a_{14} - (2 c_9
 \end{aligned} \right.
 \end{aligned} \tag{11.7}$$

$$\begin{aligned}
& -2c_{12})a_{14} + 2c_{13}a_{13} - 2c_{16}a_{24} + 2c_{18}a_{23} - 2c_{19})d_{a_{13}} + (2c_0(a_{13}a_{24} \\
& - a_{14}a_{23})a_{13} - 2c_4a_{23}a_{13} + (2c_5 + 2c_{12})a_{13} + 2c_1a_{14}a_{13} - 2c_2a_{13}^2 \\
& + 2c_3(a_{13}a_{24} - a_{14}a_{23}) + 2c_3a_{24}a_{13} + 2c_8a_{14} - 2c_{12}a_{13} + 2c_{15}a_{24} \\
& - 2c_{16}a_{23} + 2c_{17})d_{a_{14}}, (-2c_0(a_{13}a_{24} - a_{14}a_{23})a_{14} - 2c_4(a_{13}a_{24} \\
& - a_{14}a_{23}) + 2c_4a_{23}a_{14} - (2c_5 + 2c_{12})a_{14} - 2c_1a_{14}^2 + 2c_2a_{13}a_{14} \\
& - 2c_3a_{24}a_{14} - (2c_9 - 2c_{12})a_{14} + 2c_{13}a_{13} - 2c_{16}a_{24} + 2c_{18}a_{23} - 2c_{19})d_{a_{23}} \\
& + (2c_0(a_{13}a_{24} - a_{14}a_{23})a_{13} - 2c_4a_{23}a_{13} + (2c_5 + 2c_{12})a_{13} + 2c_1a_{14}a_{13} \\
& - 2c_2a_{13}^2 + 2c_3(a_{13}a_{24} - a_{14}a_{23}) + 2c_3a_{24}a_{13} + 2c_8a_{14} - 2c_{12}a_{13} \\
& + 2c_{15}a_{24} - 2c_{16}a_{23} + 2c_{17})d_{a_{24}}]
\end{aligned}$$

For the integrability condition to be satisfied by affgQ, the following differential forms should vanish on the quadric defined by affgQ:

```
> [seq(seq(Mul(Mul(Diff(affgQ, alg), Diff(alphas[i], alg), alg),
  alphas[j], alg), j=1..nops(alphas)), i=1..nops(alphas))]:
```

So we take the remainders modulo affgQ:

```
> map(eq->normal(eq/convert(alg, `*`)), %):
> Groebner[NormalForm](%, [affgQ], tdeg(op(affineVars))):
```

And we obtain polynomials in the affineVars whose coefficients should be zero.

```
> map(collect, %, affineVars, 'distributed'): map(coeffs, %,
  affineVars):
> gensIntegrability_base := map(enumer@normal, %):
```

The integrability equations do not depend on c[12].

```
> nops(select(has, gensIntegrability_base, c[12]));
```

0

(11.8)

And we do the same for all the other charts. Equivalently, we permute the coordinates of C^4 and this induces a transformations of the c's that we apply to the equations above.

```
> permutations := map[3](zip, `=`, [0,1,2,3], combinat[permute]([0,
  1,2,3]));
```

```
permutations := [[0=0, 1=1, 2=2, 3=3], [0=0, 1=1, 2=3, 3=2], [0=0, 1=2, 2
```

(11.9)

```
  =1, 3=3], [0=0, 1=2, 2=3, 3=1], [0=0, 1=3, 2=1, 3=2], [0=0, 1=3, 2
  =2, 3=1], [0=1, 1=0, 2=2, 3=3], [0=1, 1=0, 2=3, 3=2], [0=1, 1=2, 2
  =0, 3=3], [0=1, 1=2, 2=3, 3=0], [0=1, 1=3, 2=0, 3=2], [0=1, 1=3, 2
  =2, 3=0], [0=2, 1=0, 2=1, 3=3], [0=2, 1=0, 2=3, 3=1], [0=2, 1=1, 2
  =0, 3=3], [0=2, 1=1, 2=3, 3=0], [0=2, 1=3, 2=0, 3=1], [0=2, 1=3, 2
  =1, 3=0], [0=3, 1=0, 2=1, 3=2], [0=3, 1=0, 2=2, 3=1], [0=3, 1=1, 2
```

$$= 0, 3 = 2], [0 = 3, 1 = 1, 2 = 2, 3 = 0], [0 = 3, 1 = 2, 2 = 0, 3 = 1], [0 = 3, 1 = 2, 2 = 1, 3 = 0]]$$

Permutations of the coordinates of C^4 induces transformations of the Plucker coordinates:

$$\begin{aligned} & \text{> subs}([\text{seq}(\text{seq}(p[i,j] = -p[j,i], j=0..i), i=0..3)], \text{map}(p \rightarrow \text{zip}(\text{'=}', \text{pluckerVars}, \text{subs}(p, \text{pluckerVars})), \text{permutations}))]; \\ & [[p_{0,1} = p_{0,1}, p_{0,2} = p_{0,2}, p_{0,3} = p_{0,3}, p_{1,2} = p_{1,2}, p_{1,3} = p_{1,3}, p_{2,3} = p_{2,3}], [p_{0,1} \\ & = p_{0,1}, p_{0,2} = p_{0,3}, p_{0,3} = p_{0,2}, p_{1,2} = p_{1,3}, p_{1,3} = p_{1,2}, p_{2,3} = -p_{2,3}], [p_{0,1} \\ & = p_{0,2}, p_{0,2} = p_{0,1}, p_{0,3} = p_{0,3}, p_{1,2} = -p_{1,2}, p_{1,3} = p_{2,3}, p_{2,3} = p_{1,3}], [p_{0,1} \\ & = p_{0,2}, p_{0,2} = p_{0,3}, p_{0,3} = p_{0,1}, p_{1,2} = p_{2,3}, p_{1,3} = -p_{1,2}, p_{2,3} = -p_{1,3}], [p_{0,1} \\ & = p_{0,3}, p_{0,2} = p_{0,1}, p_{0,3} = p_{0,2}, p_{1,2} = -p_{1,3}, p_{1,3} = -p_{2,3}, p_{2,3} = p_{1,2}], [p_{0,1} \\ & = p_{0,3}, p_{0,2} = p_{0,2}, p_{0,3} = p_{0,1}, p_{1,2} = -p_{2,3}, p_{1,3} = -p_{1,3}, p_{2,3} = -p_{1,2}], [p_{0,1} = \\ & -p_{0,1}, p_{0,2} = p_{1,2}, p_{0,3} = p_{1,3}, p_{1,2} = p_{0,2}, p_{1,3} = p_{0,3}, p_{2,3} = p_{2,3}], [p_{0,1} = -p_{0,1}, \\ & p_{0,2} = p_{1,3}, p_{0,3} = p_{1,2}, p_{1,2} = p_{0,3}, p_{1,3} = p_{0,2}, p_{2,3} = -p_{2,3}], [p_{0,1} = p_{1,2}, p_{0,2} = \\ & -p_{0,1}, p_{0,3} = p_{1,3}, p_{1,2} = -p_{0,2}, p_{1,3} = p_{2,3}, p_{2,3} = p_{0,3}], [p_{0,1} = p_{1,2}, p_{0,2} = p_{1,3}, \\ & p_{0,3} = -p_{0,1}, p_{1,2} = p_{2,3}, p_{1,3} = -p_{0,2}, p_{2,3} = -p_{0,3}], [p_{0,1} = p_{1,3}, p_{0,2} = -p_{0,1}, \\ & p_{0,3} = p_{1,2}, p_{1,2} = -p_{0,3}, p_{1,3} = -p_{2,3}, p_{2,3} = p_{0,2}], [p_{0,1} = p_{1,3}, p_{0,2} = p_{1,2}, p_{0,3} \\ & = -p_{0,1}, p_{1,2} = -p_{2,3}, p_{1,3} = -p_{0,3}, p_{2,3} = -p_{0,2}], [p_{0,1} = -p_{0,2}, p_{0,2} = -p_{1,2}, \\ & p_{0,3} = p_{2,3}, p_{1,2} = p_{0,1}, p_{1,3} = p_{0,3}, p_{2,3} = p_{1,3}], [p_{0,1} = -p_{0,2}, p_{0,2} = p_{2,3}, p_{0,3} = \\ & -p_{1,2}, p_{1,2} = p_{0,3}, p_{1,3} = p_{0,1}, p_{2,3} = -p_{1,3}], [p_{0,1} = -p_{1,2}, p_{0,2} = -p_{0,2}, p_{0,3} \\ & = p_{2,3}, p_{1,2} = -p_{0,1}, p_{1,3} = p_{1,3}, p_{2,3} = p_{0,3}], [p_{0,1} = -p_{1,2}, p_{0,2} = p_{2,3}, p_{0,3} = \\ & -p_{0,2}, p_{1,2} = p_{1,3}, p_{1,3} = -p_{0,1}, p_{2,3} = -p_{0,3}], [p_{0,1} = p_{2,3}, p_{0,2} = -p_{0,2}, p_{0,3} = \\ & -p_{1,2}, p_{1,2} = -p_{0,3}, p_{1,3} = -p_{1,3}, p_{2,3} = p_{0,1}], [p_{0,1} = p_{2,3}, p_{0,2} = -p_{1,2}, p_{0,3} = \\ & -p_{0,2}, p_{1,2} = -p_{1,3}, p_{1,3} = -p_{0,3}, p_{2,3} = -p_{0,1}], [p_{0,1} = -p_{0,3}, p_{0,2} = -p_{1,3}, p_{0,3} \\ & = -p_{2,3}, p_{1,2} = p_{0,1}, p_{1,3} = p_{0,2}, p_{2,3} = p_{1,2}], [p_{0,1} = -p_{0,3}, p_{0,2} = -p_{2,3}, p_{0,3} = \\ & -p_{1,3}, p_{1,2} = p_{0,2}, p_{1,3} = p_{0,1}, p_{2,3} = -p_{1,2}], [p_{0,1} = -p_{1,3}, p_{0,2} = -p_{0,3}, p_{0,3} = \\ & -p_{2,3}, p_{1,2} = -p_{0,1}, p_{1,3} = p_{1,2}, p_{2,3} = p_{0,2}], [p_{0,1} = -p_{1,3}, p_{0,2} = -p_{2,3}, p_{0,3} = \\ & -p_{0,3}, p_{1,2} = p_{1,2}, p_{1,3} = -p_{0,1}, p_{2,3} = -p_{0,2}], [p_{0,1} = -p_{2,3}, p_{0,2} = -p_{0,3}, p_{0,3} = \\ & -p_{1,3}, p_{1,2} = -p_{0,2}, p_{1,3} = -p_{1,2}, p_{2,3} = p_{0,1}], [p_{0,1} = -p_{2,3}, p_{0,2} = -p_{1,3}, p_{0,3} = \\ & -p_{0,3}, p_{1,2} = -p_{1,2}, p_{1,3} = -p_{0,2}, p_{2,3} = -p_{0,1}]] \end{aligned} \quad (11.10)$$

```
> subs(c=d, map(subs, %, gQ)):
> map( p->subs(d=c,solve([coeffs(gQ - p, pluckerVars)], cvars)[1]),
%):
> map(op, map(subs, %, gensIntegrability_base)):
> gensIntegrability := [op(gbCoiso), op(convert(map(normal, %),
```


ChowFormOrSquare but not in Integrability and it is a cubic.

```
> normal(hilbIntegrability - hilbChowFormOrSquare);
t3 (11.15)
```

Let us find this polynomial:

```
> pol3 := (select(p -> degree(p) = 3, convert(gbChowFormOrSquare,
set)) minus select(p -> degree(p) = 3, convert(gbIntegrability,
set))) [1];
pol3 := -c0 c9 c20 + c1 c4 c20 + 3 c1 c10 c18 - 2 c1 c13 c17 + c2 c3 c20 + 2 c2 c8 c19
- 4 c2 c10 c16 - c3 c11 c17 - 2 c4 c6 c19 + 4 c4 c7 c17 - c5 c6 c18 + c5 c11 c15
- c7 c8 c18 - c8 c11 c16 + c9 c11 c15 (11.16)
```

```
> Groebner[Reduce](pol3, gbChowFormOrSquare, tdeg(op(cvars0)));
0 (11.17)
```

```
> Groebner[Reduce](pol3, gbIntegrability, tdeg(op(cvars0)));
-c0 c9 c20 + c1 c4 c20 + 3 c1 c10 c18 - 2 c1 c13 c17 + c2 c3 c20 + 2 c2 c8 c19 - 4 c2 c10 c16
- c3 c11 c17 - 2 c4 c6 c19 + 4 c4 c7 c17 - c5 c6 c18 + c5 c11 c15 - c7 c8 c18
- c8 c11 c16 + c9 c11 c15 (11.18)
```

```
> Groebner[Reduce](map(`*`, cvars0, pol3), gbIntegrability, tdeg(op
(cvars0)));
[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0] (11.19)
```

This concludes the proof of Proposition 3.

▼ [Annex] Save to file

Save to file the Groebner bases for future use.

```
> GBs := [gbCoiso, gbChowLines, gbChowConic, gbHurwitz, gbSquare,
gbIntegrability]:
> fd := fopen("GBs.mpl", WRITE):
> fprintf(fd, "gbCoiso := %a:\n\ngbChowLines := %a:\n\ngbChowConic
:= %a:\n\ngbHurwitz := %a:\n\ngbSquare := %a:\n\ngbIntegrability
:= %a:\n\n", op(GBs));
1242441 (12.1)
> fclose(fd):
```

Save the Groebner bases for use in Magma.

```
> fd := fopen("GBs.magma", WRITE):
> fprintf(fd, "ICoiso := Ideal( %a );\n\nIChowLines := Ideal( %a );
\n\nIChowConic := Ideal( %a );\n\nIHurwitz := Ideal( %a );
\n\nISquare := Ideal( %a );\n\nJ := Ideal( %a );\n\n",
op(subs([seq(c[i]=c||i, i=0..20)], GBs)));
840293 (12.2)
> fclose(fd):
```

