This worksheet is meant as a complement to “Computing the Chow variety of quadratic space curves”, by Peter Bürgisser, Kathlén Kohn, Pierre Lairez, and Bernd Sturmfels.

The package FGb, to compute efficiently Groebner basis, is available at http://www-polsys.lip6.fr/~jcf/FGb/FGb/index.html

```
> with(FGb): with(LinearAlgebra):
```

The computation of some of the Groebner bases require ~10 GB of RAM, so for convenienc, some of the results are preloaded.

```
> read "GBs.mpl":
```

[Section 1] Coordinates and quadrics in G(2,4)

Stiefel coordinates: a point in G(2,4) is given as the column space of a 4x2 matrix.

```
> stiefelMat := Matrix(4,2,(i,j)->a[i+10*j]);
```

$$stiefelMat := \begin{bmatrix} a_{11} & a_{21} \\ a_{12} & a_{22} \\ a_{13} & a_{23} \\ a_{14} & a_{24} \end{bmatrix} \quad \textbf{(1.1)}$$

```
> stiefelVars := [op(indets(stiefelMat))];
```

$$stiefelVars := [a_{11}, a_{12}, a_{13}, a_{14}, a_{21}, a_{22}, a_{23}, a_{24}] \quad \textbf{(1.2)}$$

(Dual) Plucker coordinates: a point in G(2,4) is given by the 2x2 minors of the matrix of Stiefel coordinates.

```
> pluckerCo := [seq(seq(p[i-1,j-1]=Determinant(stiefelMat[[i,j],1..2]), j=i+1..4), i=1..3)];
```

$$pluckerCo := [p_{0,1} = a_{11} a_{22} - a_{12} a_{21}, p_{0,2} = a_{11} a_{23} - a_{13} a_{21}, p_{0,3} = a_{11} a_{24} - a_{14} a_{21}, p_{1,2} = a_{12} a_{23} - a_{13} a_{22}, p_{1,3} = a_{12} a_{24} - a_{14} a_{22}, p_{2,3} = a_{13} a_{24} - a_{14} a_{23}] \quad \textbf{(1.3)}$$

```
> pluckerDual := { p[0,1] = p[2,3], p[0,2] = -p[1,3], p[0,3] = p[1,2], p[1,2] = p[0,3], p[1,3] = -p[0,2], p[2,3] = p[0,1] };
```

$$pluckerDual := \{p_{0,1} = p_{2,3}, p_{0,2} = -p_{1,3}, p_{0,3} = p_{1,2}, p_{1,2} = p_{0,3}, p_{1,3} = -p_{0,2}, p_{2,3} = p_{0,1}\} \quad \textbf{(1.4)}$$

```
> pluckerVars := convert(select(has,indets(pluckerCo),p),list);
```

$$pluckerVars := [p_{0,1}, p_{0,2}, p_{0,3}, p_{1,2}, p_{1,3}, p_{2,3}] \quad \textbf{(1.5)}$$

Plucker coordinates are bound by the Plucker relation

```
> pluckerRel := p[0, 1]*p[2, 3]-p[0, 2]*p[1, 3]+p[0, 3]*p[1, 2];
```

$$pluckerRel := p_{0,1} p_{2,3} - p_{0,2} p_{1,3} + p_{0,3} p_{1,2} \quad \textbf{(1.6)}$$

A generic 6x6 symmetric matrix. The change of basis c[5]=c[5]+c[12], c[9]=c[9]-c[12] makes c[12] disappear below.

```
> sym6 := subs([c[5]=c[5]+c[12], c[9]=c[9]-c[12]], Matrix(6, 6, [c
  [0], c[1], c[2], c[3], c[4], c[5], c[1], c[6], c[7], c[8], c[9],
  c[10], c[2], c[7], c[11], c[12], c[13], c[14], c[3], c[8], c[12],
  c[15], c[16], c[17], c[4], c[9], c[13], c[16], c[18], c[19], c
  [5], c[10], c[14], c[17], c[19], c[20]]));
```

$$sym6 := \begin{bmatrix} c_0 & c_1 & c_2 & c_3 & c_4 & c_5 + c_{12} \\ c_1 & c_6 & c_7 & c_8 & c_9 - c_{12} & c_{10} \\ c_2 & c_7 & c_{11} & c_{12} & c_{13} & c_{14} \\ c_3 & c_8 & c_{12} & c_{15} & c_{16} & c_{17} \\ c_4 & c_9 - c_{12} & c_{13} & c_{16} & c_{18} & c_{19} \\ c_5 + c_{12} & c_{10} & c_{14} & c_{17} & c_{19} & c_{20} \end{bmatrix} \tag{1.7}$$

```
> cvars := [seq(c[i], i=0..20)];
```

$$cvars := [c_0, c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_9, c_{10}, c_{11}, c_{12}, c_{13}, c_{14}, c_{15}, c_{16}, c_{17}, c_{18}, c_{19}, c_{20}] \tag{1.8}$$

The ideals we will compute do not depend on c[12], as explain in the article.

```
> cvars0 := remove(`=`, cvars, c[12]);
```

$$cvars0 := [c_0, c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_9, c_{10}, c_{11}, c_{13}, c_{14}, c_{15}, c_{16}, c_{17}, c_{18}, c_{19}, c_{20}] \tag{1.9}$$

A generic quadratic form in the Plucker coordinates.

```
> gQ := collect(Vector[row](pluckerVars).sym6.Vector(pluckerVars),
  pluckerVars, distributed);
```

$$\begin{aligned} gQ := {} & c_0 p_{0,1}^2 + 2 c_4 p_{1,3} p_{0,1} + (2 c_5 + 2 c_{12}) p_{2,3} p_{0,1} + 2 c_1 p_{0,2} p_{0,1} \\ & + 2 c_2 p_{0,3} p_{0,1} + 2 c_3 p_{1,2} p_{0,1} + c_6 p_{0,2}^2 + (2 c_9 - 2 c_{12}) p_{1,3} p_{0,2} \\ & + 2 c_{10} p_{2,3} p_{0,2} + 2 c_7 p_{0,3} p_{0,2} + 2 c_8 p_{1,2} p_{0,2} + c_{11} p_{0,3}^2 + 2 c_{12} p_{1,2} p_{0,3} \\ & + 2 c_{13} p_{1,3} p_{0,3} + 2 c_{14} p_{2,3} p_{0,3} + c_{15} p_{1,2}^2 + 2 c_{16} p_{1,3} p_{1,2} + 2 c_{17} p_{2,3} p_{1,2} \\ & + c_{18} p_{1,3}^2 + 2 p_{2,3} c_{19} p_{1,3} + p_{2,3}^2 c_{20} \end{aligned} \tag{1.10}$$

A quadratic form in the Plucker coordinates defines a quadric hypersurface of G(2,4). Two forms that differ by a multiple of the Plucker relation give the same quadric. Thus, the set of quadric of G(2,4) can be identified with the projectivization of the quotient of C^21 (the quadratic forms) by the Plucker relation (cf. the article).

[Section 2] Ideal of coisotropic quadrics in G(2,4)

Now, let us see the condition that gQ defines a coisotropic hypersurface.
The quadric defined by gQ is coisotropic if there are t[0], t[1] and t[2], not all zero, such that the following vanishes:

```
> coisoCondition := t[0]*(diff(gQ, p[0,1])*diff(gQ,p[2,3])-diff(gQ,
  p[0,2])*diff(gQ,p[1,3])+diff(gQ,p[0,3])*diff(gQ,p[1,2])) + t[1]*
  gQ + t[2]*pluckerRel:
```

So here are the equations:

```
> coisoConditionCoeffs := [coeffs(collect(coisoCondition,
  pluckerVars, 'distributed'), pluckerVars)];
```

$$
\begin{aligned}
\mathit{coisoConditionCoeffs} := \Big[& 2\, t_1\, c_{10} + t_0 \left(2 \left(2\, c_5 + 2\, c_{12}\right) c_{10} + 4\, c_1\, c_{20} + 4\, c_{14}\, c_8\right.\\
& \left. + 4\, c_7\, c_{17} - 2\, c_{10} \left(2\, c_9 - 2\, c_{12}\right) - 4\, c_6\, c_{19}\right), t_0 \left(2 \left(2\, c_5 + 2\, c_{12}\right) c_{14} + 4\, c_2\, c_{20}\right.\\
& \left. - 4\, c_{10}\, c_{13} - 4\, c_7\, c_{19} + 4\, c_{14}\, c_{12} + 4\, c_{11}\, c_{17}\right) + 2\, t_1\, c_{14}, t_0 \left(2 \left(2\, c_5 + 2\, c_{12}\right) c_{17}\right.\\
& \left. + 4\, c_3\, c_{20} + 4\, c_{14}\, c_{15} + 4\, c_{12}\, c_{17} - 4\, c_{10}\, c_{16} - 4\, c_8\, c_{19}\right) + 2\, t_1\, c_{17}, t_0 \left(2 \left(2\, c_5\right.\right.\\
& \left.\left. + 2\, c_{12}\right) c_{19} + 4\, c_4\, c_{20} - 4\, c_{10}\, c_{18} - 2 \left(2\, c_9 - 2\, c_{12}\right) c_{19} + 4\, c_{13}\, c_{17} + 4\, c_{14}\, c_{16}\right)\\
& + 2\, t_1\, c_{19}, t_0 \left(4\, c_4\, c_{17} + 4\, c_3\, c_{19} + 4\, c_{13}\, c_{15} + 4\, c_{12}\, c_{16} - 2 \left(2\, c_9 - 2\, c_{12}\right) c_{16}\right.\\
& \left. - 4\, c_8\, c_{18}\right) + 2\, t_1\, c_{16}, t_0 \left(\left(2\, c_5 + 2\, c_{12}\right)^2 + 4\, c_0\, c_{20} - 4\, c_{10}\, c_4 - 4\, c_1\, c_{19} + 4\, c_{14}\, c_3\right.\\
& \left. + 4\, c_2\, c_{17}\right) + t_1 \left(2\, c_5 + 2\, c_{12}\right) + t_2, t_1 \left(2\, c_9 - 2\, c_{12}\right) - t_2 + t_0 \left(4\, c_{10}\, c_4 + 4\, c_1\, c_{19}\right.\\
& \left. + 4\, c_{13}\, c_8 + 4\, c_7\, c_{16} - \left(2\, c_9 - 2\, c_{12}\right)^2 - 4\, c_6\, c_{18}\right), t_0 \left(4\, c_2\, c_{17} + 4\, c_3\, c_{14} - 4\, c_7\, c_{16}\right.\\
& \left. - 4\, c_8\, c_{13} + 4\, c_{11}\, c_{15} + 4\, c_{12}^2\right) + t_2 + 2\, t_1\, c_{12}, t_0 \left(4\, c_4\, c_{14} + 4\, c_2\, c_{19} - 2 \left(2\, c_9\right.\right.\\
& \left.\left. - 2\, c_{12}\right) c_{13} - 4\, c_7\, c_{18} + 4\, c_{13}\, c_{12} + 4\, c_{11}\, c_{16}\right) + 2\, t_1\, c_{13}, 2\, t_1\, c_8 + t_0 \left(4\, c_3\, c_{10}\right.\\
& \left. + 4\, c_1\, c_{17} + 4\, c_{12}\, c_8 + 4\, c_7\, c_{15} - 2\, c_8 \left(2\, c_9 - 2\, c_{12}\right) - 4\, c_6\, c_{16}\right), 2\, t_1\, c_7\\
& + t_0 \left(4\, c_2\, c_{10} + 4\, c_1\, c_{14} + 4\, c_{11}\, c_8 + 4\, c_7\, c_{12} - 2\, c_7 \left(2\, c_9 - 2\, c_{12}\right) - 4\, c_6\, c_{13}\right),\\
& t_0 \left(2\, c_4 \left(2\, c_5 + 2\, c_{12}\right) + 4\, c_0\, c_{19} - 2 \left(2\, c_9 - 2\, c_{12}\right) c_4 - 4\, c_1\, c_{18} + 4\, c_{13}\, c_3\right.\\
& \left. + 4\, c_2\, c_{16}\right) + 2\, t_1\, c_4, t_0 \left(2\, c_3 \left(2\, c_5 + 2\, c_{12}\right) + 4\, c_0\, c_{17} - 4\, c_8\, c_4 - 4\, c_1\, c_{16}\right.\\
& \left. + 4\, c_{12}\, c_3 + 4\, c_2\, c_{15}\right) + 2\, t_1\, c_3, t_0 \left(2\, c_2 \left(2\, c_5 + 2\, c_{12}\right) + 4\, c_0\, c_{14} - 4\, c_7\, c_4\right.\\
& \left. - 4\, c_1\, c_{13} + 4\, c_{11}\, c_3 + 4\, c_2\, c_{12}\right) + 2\, t_1\, c_2, t_0 \left(2\, c_1 \left(2\, c_5 + 2\, c_{12}\right) + 4\, c_0\, c_{10}\right.\\
& \left. - 4\, c_6\, c_4 - 2\, c_1 \left(2\, c_9 - 2\, c_{12}\right) + 4\, c_7\, c_3 + 4\, c_2\, c_8\right) + 2\, t_1\, c_1, t_0 \left(2 \left(2\, c_5\right.\right.\\
& \left.\left. + 2\, c_{12}\right) c_{20} + 4\, c_{14}\, c_{17} - 4\, c_{10}\, c_{19}\right) + t_1\, c_{20}, t_0 \left(4\, c_3\, c_{17} - 4\, c_8\, c_{16} + 4\, c_{12}\, c_{15}\right)\\
& + t_1\, c_{15}, t_0 \left(-2 \left(2\, c_9 - 2\, c_{12}\right) c_{18} + 4\, c_4\, c_{19} + 4\, c_{13}\, c_{16}\right) + t_1\, c_{18}, t_0 \left(4\, c_7\, c_8\right.\\
& \left. + 4\, c_1\, c_{10} - 2\, c_6 \left(2\, c_9 - 2\, c_{12}\right)\right) + t_1\, c_6, t_1\, c_{11} + t_0 \left(4\, c_2\, c_{14} - 4\, c_7\, c_{13}\right.\\
& \left. + 4\, c_{11}\, c_{12}\right), t_1\, c_0 + t_0 \left(-4\, c_1\, c_4 + 2\, c_0 \left(2\, c_5 + 2\, c_{12}\right) + 4\, c_2\, c_3\right) \Big]
\end{aligned}
\tag{2.1}
$$

The t[i]'s are eliminated using linear algebra.

```
> coisoConditionMat := Matrix(map(e -> [seq(coeff(e, t[i]), i=0..2)
  ], coisoConditionCoeffs));
```

$$
\mathit{coisoConditionMat} := \begin{bmatrix} \textit{21 x 3 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix}
\tag{2.2}
$$

The equations we are looking for are the 3x3 minors of the matrix above.

```
> gensCoiso := convert(convert([seq(Determinant(coisoConditionMat
  [p,1..3]), p in combinat[choose]([seq(i, i in 1..21)],3))], set),
  list):
```

The equations do not depend on c[12].

```
> nops(select(has, gensCoiso, c[12]));
```

$$0 \tag{2.3}$$

And we compute a Groebner basis of the ideal generated by these equations. (Preloaded.)

```
> # gbCoiso := fgb_gbasis(gensCoiso, 0, cvars, []):
```

[Section 3] Ideal of the Chow forms of two lines

We start with one line given by the Plucker coordinates q[i,j]

The line given by the p[i,j] meets the other if and only if the following vanishes:

```
> meetCondition := p[0,1]*q[2,3] - p[0,2]*q[1,3] + p[0,3]*q[1,2] +
  p[2,3]*q[0,1] - p[1,3]*q[0,2] + p[1,2]*q[0,3];
```

$$meetCondition := p_{0,1}\, q_{2,3} - p_{0,2}\, q_{1,3} + p_{0,3}\, q_{1,2} + p_{1,2}\, q_{0,3} - p_{1,3}\, q_{0,2} + p_{2,3}\, q_{0,1} \tag{3.1}$$

Indeed, translated into Stiefel coordinates, this polynomial is just a determinant:

```
> Matrix([stiefelMat, subs(a=b,stiefelMat)]);
```

$$\begin{bmatrix} a_{11} & a_{21} & b_{11} & b_{21} \\ a_{12} & a_{22} & b_{12} & b_{22} \\ a_{13} & a_{23} & b_{13} & b_{23} \\ a_{14} & a_{24} & b_{14} & b_{24} \end{bmatrix} \tag{3.2}$$

```
> normal(subs(pluckerCo, subs([p=q,a=b], pluckerCo), meetCondition)
  - Determinant(%));
```

$$0 \tag{3.3}$$

Therefore, the quadric gQ is the Chow form of a pair of skew lines if it equals the product of form meetCondition above, for some lines q[i,j] and s[i,j], modulo the Plucker relation.

```
> Groebner[Reduce](gQ - meetCondition*subs(q=s, meetCondition),
  [pluckerRel], tdeg(op(pluckerVars)));
```

$$\begin{aligned} &\left(-q_{2,3}\, s_{2,3} + c_0\right) p_{0,1}^2 + \left(-q_{0,1}\, s_{2,3} + q_{0,3}\, s_{1,2} + q_{1,2}\, s_{0,3} - q_{2,3}\, s_{0,1} \right. \\ &\quad \left. + 2\, c_5\right) p_{2,3}\, p_{0,1} + \left(-q_{0,2}\, s_{1,3} - q_{0,3}\, s_{1,2} - q_{1,2}\, s_{0,3} - q_{1,3}\, s_{0,2} \right. \\ &\quad \left. + 2\, c_9\right) p_{1,3}\, p_{0,2} + \left(q_{0,1}\, s_{1,3} + q_{1,3}\, s_{0,1} + 2\, c_{10}\right) p_{0,2}\, p_{2,3} + \left(-q_{0,1}\, s_{1,2} \right. \\ &\quad \left. - q_{1,2}\, s_{0,1} + 2\, c_{14}\right) p_{0,3}\, p_{2,3} + \left(-q_{0,1}\, s_{0,3} - q_{0,3}\, s_{0,1} + 2\, c_{17}\right) p_{1,2}\, p_{2,3} \\ &\quad + \left(q_{0,1}\, s_{0,2} + q_{0,2}\, s_{0,1} + 2\, c_{19}\right) p_{1,3}\, p_{2,3} + \left(q_{0,3}\, s_{1,3} + q_{1,3}\, s_{0,3} \right. \\ &\quad \left. + 2\, c_8\right) p_{0,2}\, p_{1,2} + \left(q_{0,2}\, s_{2,3} + q_{2,3}\, s_{0,2} + 2\, c_4\right) p_{0,1}\, p_{1,3} + \left(q_{0,2}\, s_{1,2} \right. \\ &\quad \left. + q_{1,2}\, s_{0,2} + 2\, c_{13}\right) p_{0,3}\, p_{1,3} + \left(q_{0,2}\, s_{0,3} + q_{0,3}\, s_{0,2} + 2\, c_{16}\right) p_{1,2}\, p_{1,3} \\ &\quad + \left(q_{1,3}\, s_{2,3} + q_{2,3}\, s_{1,3} + 2\, c_1\right) p_{0,1}\, p_{0,2} + \left(-q_{1,2}\, s_{2,3} - q_{2,3}\, s_{1,2} \right. \end{aligned} \tag{3.4}$$

$+ 2\,c_2)\,p_{0,1}\,p_{0,3} + (q_{1,2}\,s_{1,3} + q_{1,3}\,s_{1,2} + 2\,c_7)\,p_{0,2}\,p_{0,3} + (-q_{0,3}\,s_{2,3} - q_{2,3}\,s_{0,3} + 2\,c_3)\,p_{0,1}\,p_{1,2} + (-q_{0,1}\,s_{0,1} + c_{20})\,p_{2,3}^2 + (-q_{1,3}\,s_{1,3} + c_6)\,p_{0,2}^2 + (-q_{0,3}\,s_{0,3} + c_{15})\,p_{1,2}^2 + (-q_{0,2}\,s_{0,2} + c_{18})\,p_{1,3}^2 + (-q_{1,2}\,s_{1,2} + c_{11})\,p_{0,3}^2$

```
> eqsChowLines := [coeffs(%, pluckerVars), subs(p=q, pluckerRel),
  subs(p=s, pluckerRel)];
```

$$eqsChowLines := [-q_{2,3}\,s_{2,3} + c_0,\ -q_{0,1}\,s_{2,3} + q_{0,3}\,s_{1,2} + q_{1,2}\,s_{0,3} - q_{2,3}\,s_{0,1} + 2\,c_5,\ -q_{0,2}\,s_{1,3} - q_{0,3}\,s_{1,2} - q_{1,2}\,s_{0,3} - q_{1,3}\,s_{0,2} + 2\,c_9,\ q_{0,1}\,s_{1,3} + q_{1,3}\,s_{0,1} + 2\,c_{10},\ -q_{0,1}\,s_{1,2} - q_{1,2}\,s_{0,1} + 2\,c_{14},\ -q_{0,1}\,s_{0,3} - q_{0,3}\,s_{0,1} + 2\,c_{17},\ q_{0,1}\,s_{0,2} + q_{0,2}\,s_{0,1} + 2\,c_{19},\ q_{0,3}\,s_{1,3} + q_{1,3}\,s_{0,3} + 2\,c_8,\ q_{0,2}\,s_{2,3} + q_{2,3}\,s_{0,2} + 2\,c_4,\ q_{0,2}\,s_{1,2} + q_{1,2}\,s_{0,2} + 2\,c_{13},\ q_{0,2}\,s_{0,3} + q_{0,3}\,s_{0,2} + 2\,c_{16},\ q_{1,3}\,s_{2,3} + q_{2,3}\,s_{1,3} + 2\,c_1,\ -q_{1,2}\,s_{2,3} - q_{2,3}\,s_{1,2} + 2\,c_2,\ q_{1,2}\,s_{1,3} + q_{1,3}\,s_{1,2} + 2\,c_7,\ -q_{0,3}\,s_{2,3} - q_{2,3}\,s_{0,3} + 2\,c_3,\ -q_{0,1}\,s_{0,1} + c_{20},\ -q_{1,3}\,s_{1,3} + c_6,\ -q_{0,3}\,s_{0,3} + c_{15},\ -q_{0,2}\,s_{0,2} + c_{18},\ -q_{1,2}\,s_{1,2} + c_{11},\ q_{0,1}\,q_{2,3} - q_{0,2}\,q_{1,3} + q_{0,3}\,q_{1,2},\ s_{0,1}\,s_{2,3} - s_{0,2}\,s_{1,3} + s_{0,3}\,s_{1,2}] \quad (3.5)$$

Again, the equations do not depend on c[12].

```
> nops(select(has, eqsChowLines, c[12]));
```

$$0 \quad (3.6)$$

It only remains to eleminate the q's and the s's.

```
> [op(indets(eqsChowLines) minus {op(cvars)})], cvars;
```

$$[q_{0,1}, q_{0,2}, q_{0,3}, q_{1,2}, q_{1,3}, q_{2,3}, s_{0,1}, s_{0,2}, s_{0,3}, s_{1,2}, s_{1,3}, s_{2,3}],\ [c_0, c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_9, c_{10}, c_{11}, c_{12}, c_{13}, c_{14}, c_{15}, c_{16}, c_{17}, c_{18}, c_{19}, c_{20}] \quad (3.7)$$

WARNING : the computation below requires ~ 10GB of RAM.

```
> # gbChowLines := fgb_gbasis_elim(eqsChowLines, 0, %, {"verb" = 3,
  "index" = 10^8}):
```

[Section 4] Ideal of the Chow forms of plane conics

Let us consider a plane conic in P^3, given by a linear form f1 and a quadratic form f2.

```
> pX := stiefelMat.Vector([1,t]);
```

$$pX := \begin{bmatrix} t\,a_{21} + a_{11} \\ t\,a_{22} + a_{12} \\ t\,a_{23} + a_{13} \\ t\,a_{24} + a_{14} \end{bmatrix} \quad (4.1)$$

```
> f1 := collect(add(u[i]*pX[i], i=1..4), t, normal);
```

(4.2)

$$f1 := \left(a_{21}\,u_1 + a_{22}\,u_2 + a_{23}\,u_3 + a_{24}\,u_4\right) t + u_1\,a_{11} + u_2\,a_{12} + u_3\,a_{13} + u_4\,a_{14} \tag{4.2}$$

```
> f2 := collect(Transpose(pX).Matrix(4,4,(i,j) -> v[i+10*j], shape=
  symmetric).pX, t, normal);
```

$$\begin{aligned}
f2 := \Big(& a_{21}^2\,v_{11} + 2\,a_{21}\,a_{22}\,v_{21} + 2\,a_{21}\,a_{23}\,v_{31} + 2\,a_{21}\,a_{24}\,v_{41} + a_{22}^2\,v_{22} + 2\,a_{22}\,a_{23}\,v_{32} \\
& + 2\,a_{22}\,a_{24}\,v_{42} + a_{23}^2\,v_{33} + 2\,a_{23}\,a_{24}\,v_{43} + a_{24}^2\,v_{44}\Big)\,t^2 + \Big(2\,a_{11}\,a_{21}\,v_{11} \\
& + 2\,a_{11}\,a_{22}\,v_{21} + 2\,a_{11}\,a_{23}\,v_{31} + 2\,a_{11}\,a_{24}\,v_{41} + 2\,a_{12}\,a_{21}\,v_{21} + 2\,a_{12}\,a_{22}\,v_{22} \\
& + 2\,a_{12}\,a_{23}\,v_{32} + 2\,a_{12}\,a_{24}\,v_{42} + 2\,a_{13}\,a_{21}\,v_{31} + 2\,a_{13}\,a_{22}\,v_{32} + 2\,a_{13}\,a_{23}\,v_{33} \\
& + 2\,a_{13}\,a_{24}\,v_{43} + 2\,a_{14}\,a_{21}\,v_{41} + 2\,a_{14}\,a_{22}\,v_{42} + 2\,a_{14}\,a_{23}\,v_{43} + 2\,a_{14}\,a_{24}\,v_{44}\Big)\,t + \\
& a_{11}^2\,v_{11} + 2\,a_{11}\,a_{12}\,v_{21} + 2\,a_{11}\,a_{13}\,v_{31} + 2\,a_{11}\,a_{14}\,v_{41} + a_{12}^2\,v_{22} + 2\,a_{12}\,a_{13}\,v_{32} \\
& + 2\,a_{12}\,a_{14}\,v_{42} + a_{13}^2\,v_{33} + 2\,a_{13}\,a_{14}\,v_{43} + a_{14}^2\,v_{44}
\end{aligned} \tag{4.3}$$

A line in P^3, given by Stiefel coordinates meets the conic if and only if the following resultant vanishes:

```
> res := resultant(f1, f2, t):
```

We can express this resultant as a quadratic form in the Plucker coordinates.

```
> pluckerRes := fgb_gbasis_elim([res, seq(pluckerVars[i]-subs
  (pluckerCo,pluckerVars[i]), i=1..6)], 0, stiefelVars, [op(indets
  (res) minus {op(stiefelVars)}), op(pluckerVars)])[1];
```

$$\begin{aligned}
pluckerRes := {} & p_{0,1}^2\,u_1^2\,v_{22} - 2\,p_{0,1}^2\,u_1\,u_2\,v_{21} + p_{0,1}^2\,u_2^2\,v_{11} + 2\,p_{0,1}\,p_{0,2}\,u_1^2\,v_{32} \\
& - 2\,p_{0,1}\,p_{0,2}\,u_1\,u_2\,v_{31} - 2\,p_{0,1}\,p_{0,2}\,u_1\,u_3\,v_{21} + 2\,p_{0,1}\,p_{0,2}\,u_2\,u_3\,v_{11} + 2\,p_{0,1}\,p_{0,3} \\
& u_1^2\,v_{42} - 2\,p_{0,1}\,p_{0,3}\,u_1\,u_2\,v_{41} - 2\,p_{0,1}\,p_{0,3}\,u_1\,u_4\,v_{21} + 2\,p_{0,1}\,p_{0,3}\,u_2\,u_4\,v_{11} \\
& + 2\,p_{0,1}\,p_{1,2}\,u_1\,u_2\,v_{32} - 2\,p_{0,1}\,p_{1,2}\,u_1\,u_3\,v_{22} - 2\,p_{0,1}\,p_{1,2}\,u_2^2\,v_{31} \\
& + 2\,p_{0,1}\,p_{1,2}\,u_2\,u_3\,v_{21} + 2\,p_{0,1}\,p_{1,3}\,u_1\,u_2\,v_{42} - 2\,p_{0,1}\,p_{1,3}\,u_1\,u_4\,v_{22} - 2\,p_{0,1}\,p_{1,3} \\
& u_2^2\,v_{41} + 2\,p_{0,1}\,p_{1,3}\,u_2\,u_4\,v_{21} - 2\,p_{0,1}\,p_{2,3}\,u_1\,u_2\,v_{43} + 4\,p_{0,1}\,p_{2,3}\,u_1\,u_3\,v_{42} \\
& - 2\,p_{0,1}\,p_{2,3}\,u_1\,u_4\,v_{32} - 2\,p_{0,1}\,p_{2,3}\,u_2\,u_3\,v_{41} + 4\,p_{0,1}\,p_{2,3}\,u_2\,u_4\,v_{31} \\
& - 2\,p_{0,1}\,p_{2,3}\,u_3\,u_4\,v_{21} + p_{0,2}^2\,u_1^2\,v_{33} - 2\,p_{0,2}^2\,u_1\,u_3\,v_{31} + p_{0,2}^2\,u_3^2\,v_{11} + 2\,p_{0,2}\,p_{0,3} \\
& u_1^2\,v_{43} - 2\,p_{0,2}\,p_{0,3}\,u_1\,u_3\,v_{41} - 2\,p_{0,2}\,p_{0,3}\,u_1\,u_4\,v_{31} + 2\,p_{0,2}\,p_{0,3}\,u_3\,u_4\,v_{11} \\
& + 2\,p_{0,2}\,p_{1,2}\,u_1\,u_2\,v_{33} - 2\,p_{0,2}\,p_{1,2}\,u_1\,u_3\,v_{32} - 2\,p_{0,2}\,p_{1,2}\,u_2\,u_3\,v_{31} + 2\,p_{0,2}\,p_{1,2} \\
& u_3^2\,v_{21} + 4\,p_{0,2}\,p_{1,3}\,u_1\,u_2\,v_{43} - 2\,p_{0,2}\,p_{1,3}\,u_1\,u_3\,v_{42} - 2\,p_{0,2}\,p_{1,3}\,u_1\,u_4\,v_{32} \\
& - 2\,p_{0,2}\,p_{1,3}\,u_2\,u_3\,v_{41} - 2\,p_{0,2}\,p_{1,3}\,u_2\,u_4\,v_{31} + 4\,p_{0,2}\,p_{1,3}\,u_3\,u_4\,v_{21} \\
& + 2\,p_{0,2}\,p_{2,3}\,u_1\,u_3\,v_{43} - 2\,p_{0,2}\,p_{2,3}\,u_1\,u_4\,v_{33} - 2\,p_{0,2}\,p_{2,3}\,u_3^2\,v_{41} \\
& + 2\,p_{0,2}\,p_{2,3}\,u_3\,u_4\,v_{31} + p_{0,3}^2\,u_1^2\,v_{44} - 2\,p_{0,3}^2\,u_1\,u_4\,v_{41} + p_{0,3}^2\,u_4^2\,v_{11} \\
& + 2\,p_{0,3}\,p_{1,3}\,u_1\,u_2\,v_{44} - 2\,p_{0,3}\,p_{1,3}\,u_1\,u_4\,v_{42} - 2\,p_{0,3}\,p_{1,3}\,u_2\,u_4\,v_{41} + 2\,p_{0,3}\,p_{1,3}
\end{aligned} \tag{4.4}$$

$u_4^2 v_{21} + 2 p_{0,3} p_{2,3} u_1 u_3 v_{44} - 2 p_{0,3} p_{2,3} u_1 u_4 v_{43} - 2 p_{0,3} p_{2,3} u_3 u_4 v_{41} + 2 p_{0,3} p_{2,3} u_4^2 v_{31} + p_{1,2}^2 u_2^2 v_{33} - 2 p_{1,2}^2 u_2 u_3 v_{32} + p_{1,2}^2 u_3^2 v_{22} + 2 p_{1,2} p_{1,3} u_2^2 v_{43} - 2 p_{1,2} p_{1,3} u_2 u_3 v_{42} - 2 p_{1,2} p_{1,3} u_2 u_4 v_{32} + 2 p_{1,2} p_{1,3} u_3 u_4 v_{22} + 2 p_{1,2} p_{2,3} u_2 u_3 v_{43} - 2 p_{1,2} p_{2,3} u_2 u_4 v_{33} - 2 p_{1,2} p_{2,3} u_3^2 v_{42} + 2 p_{1,2} p_{2,3} u_3 u_4 v_{32} + p_{1,3}^2 u_2^2 v_{44} - 2 p_{1,3}^2 u_2 u_4 v_{42} + p_{1,3}^2 u_4^2 v_{22} + 2 p_{1,3} p_{2,3} u_2 u_3 v_{44} - 2 p_{1,3} p_{2,3} u_2 u_4 v_{43} - 2 p_{1,3} p_{2,3} u_3 u_4 v_{42} + 2 p_{1,3} p_{2,3} u_4^2 v_{32} + p_{2,3}^2 u_3^2 v_{44} - 2 p_{2,3}^2 u_3 u_4 v_{43} + p_{2,3}^2 u_4^2 v_{33}$

Therefore, the quadric gQ is the Chow form of a plane conic if it equals the form above, for some f1 and f2, modulo the Plucker relation.

```
> Groebner[NormalForm](gQ - pluckerRes, [pluckerRel], tdeg(op
  (pluckerVars)));
```

$$\left(-2 u_2 u_3 v_{43} + 2 u_2 u_4 v_{33} + 2 u_3^2 v_{42} - 2 u_3 u_4 v_{32} + 2 c_{17}\right) p_{1,2} p_{2,3} + \left(-2 u_2 u_3 v_{44} + 2 u_2 u_4 v_{43} + 2 u_3 u_4 v_{42} - 2 u_4^2 v_{32} + 2 c_{19}\right) p_{1,3} p_{2,3} + \left(-2 u_1^2 v_{43} + 2 u_1 u_3 v_{41} + 2 u_1 u_4 v_{31} - 2 u_3 u_4 v_{11} + 2 c_7\right) p_{0,2} p_{0,3} + \left(-2 u_1 u_2 v_{32} + 2 u_1 u_3 v_{22} + 2 u_2^2 v_{31} - 2 u_2 u_3 v_{21} + 2 c_3\right) p_{0,1} p_{1,2} + \left(-2 u_1 u_2 v_{33} + 2 u_1 u_3 v_{32} + 2 u_2 u_3 v_{31} - 2 u_3^2 v_{21} + 2 c_8\right) p_{0,2} p_{1,2} + \left(-2 u_1 u_2 v_{42} + 2 u_1 u_4 v_{22} + 2 u_2^2 v_{41} - 2 u_2 u_4 v_{21} + 2 c_4\right) p_{0,1} p_{1,3} + \left(-2 u_1 u_2 v_{44} + 2 u_1 u_4 v_{42} + 2 u_2 u_4 v_{41} - 2 u_4^2 v_{21} + 2 c_{13}\right) p_{0,3} p_{1,3} + \left(-2 u_2^2 v_{43} + 2 u_2 u_3 v_{42} + 2 u_2 u_4 v_{32} - 2 u_3 u_4 v_{22} + 2 c_{16}\right) p_{1,2} p_{1,3} + \left(-2 u_1 u_3 v_{43} + 2 u_1 u_4 v_{33} + 2 u_3^2 v_{41} - 2 u_3 u_4 v_{31} + 2 c_{10}\right) p_{0,2} p_{2,3} + \left(-2 u_1 u_3 v_{44} + 2 u_1 u_4 v_{43} + 2 u_3 u_4 v_{41} - 2 u_4^2 v_{31} + 2 c_{14}\right) p_{0,3} p_{2,3} + \left(-u_2^2 v_{44} + 2 u_2 u_4 v_{42} - u_4^2 v_{22} + c_{18}\right) p_{1,3}^2 + \left(-u_1^2 v_{33} + 2 u_1 u_3 v_{31} - u_3^2 v_{11} + c_6\right) p_{0,2}^2 + \left(-u_1^2 v_{44} + 2 u_1 u_4 v_{41} - u_4^2 v_{11} + c_{11}\right) p_{0,3}^2 + \left(-u_2^2 v_{33} + 2 u_2 u_3 v_{32} - u_3^2 v_{22} + c_{15}\right) p_{1,2}^2 + \left(-u_1^2 v_{22} + 2 u_1 u_2 v_{21} - u_2^2 v_{11} + c_0\right) p_{0,1}^2 + \left(2 u_1 u_2 v_{43} - 4 u_1 u_3 v_{42} + 2 u_1 u_4 v_{32} + 2 u_2 u_3 v_{41} - 4 u_2 u_4 v_{31} + 2 u_3 u_4 v_{21} + 2 c_5\right) p_{0,1} p_{2,3} + \left(-4 u_1 u_2 v_{43} + 2 u_1 u_3 v_{42} + 2 u_1 u_4 v_{32} + 2 u_2 u_3 v_{41} + 2 u_2 u_4 v_{31} - 4 u_3 u_4 v_{21} + 2 c_9\right) p_{0,2} p_{1,3} + \left(-2 u_1^2 v_{32} + 2 u_1 u_2 v_{31} + 2 u_1 u_3 v_{21} - 2 u_2 u_3 v_{11} + 2 c_1\right) p_{0,1} p_{0,2} + \left(-2 u_1^2 v_{42} + 2 u_1 u_2 v_{41} + 2 u_1 u_4 v_{21} - 2 u_2 u_4 v_{11} + 2 c_2\right) p_{0,1} p_{0,3} + \left(-u_3^2 v_{44} + 2 u_3 u_4 v_{43} - u_4^2 v_{33} + c_{20}\right) p_{2,3}^2 \qquad \textbf{(4.5)}$$

```
> eqsChowConic := [coeffs(%, pluckerVars)];
```

$$eqsChowConic := \left[-2\,u_2\,u_3\,v_{43} + 2\,u_2\,u_4\,v_{33} + 2\,u_3^2\,v_{42} - 2\,u_3\,u_4\,v_{32} + 2\,c_{17}, -2\,u_2\,u_3\,v_{44} + 2\,u_2\,u_4\,v_{43} + 2\,u_3\,u_4\,v_{42} - 2\,u_4^2\,v_{32} + 2\,c_{19}, -2\,u_1^2\,v_{43} + 2\,u_1\,u_3\,v_{41} + 2\,u_1\,u_4\,v_{31} - 2\,u_3\,u_4\,v_{11} + 2\,c_7, -2\,u_1\,u_2\,v_{32} + 2\,u_1\,u_3\,v_{22} + 2\,u_2^2\,v_{31} - 2\,u_2\,u_3\,v_{21} + 2\,c_3, -2\,u_1\,u_2\,v_{33} + 2\,u_1\,u_3\,v_{32} + 2\,u_2\,u_3\,v_{31} - 2\,u_3^2\,v_{21} + 2\,c_8, -2\,u_1\,u_2\,v_{42} + 2\,u_1\,u_4\,v_{22} + 2\,u_2^2\,v_{41} - 2\,u_2\,u_4\,v_{21} + 2\,c_4, -2\,u_1\,u_2\,v_{44} + 2\,u_1\,u_4\,v_{42} + 2\,u_2\,u_4\,v_{41} - 2\,u_4^2\,v_{21} + 2\,c_{13}, -2\,u_2^2\,v_{43} + 2\,u_2\,u_3\,v_{42} + 2\,u_2\,u_4\,v_{32} - 2\,u_3\,u_4\,v_{22} + 2\,c_{16}, -2\,u_1\,u_3\,v_{43} + 2\,u_1\,u_4\,v_{33} + 2\,u_3^2\,v_{41} - 2\,u_3\,u_4\,v_{31} + 2\,c_{10}, -2\,u_1\,u_3\,v_{44} + 2\,u_1\,u_4\,v_{43} + 2\,u_3\,u_4\,v_{41} - 2\,u_4^2\,v_{31} + 2\,c_{14}, -u_2^2\,v_{44} + 2\,u_2\,u_4\,v_{42} - u_4^2\,v_{22} + c_{18}, -u_1^2\,v_{33} + 2\,u_1\,u_3\,v_{31} - u_3^2\,v_{11} + c_6, -u_1^2\,v_{44} + 2\,u_1\,u_4\,v_{41} - u_4^2\,v_{11} + c_{11}, -u_2^2\,v_{33} + 2\,u_2\,u_3\,v_{32} - u_3^2\,v_{22} + c_{15}, -u_1^2\,v_{22} + 2\,u_1\,u_2\,v_{21} - u_2^2\,v_{11} + c_0, 2\,u_1\,u_2\,v_{43} - 4\,u_1\,u_3\,v_{42} + 2\,u_1\,u_4\,v_{32} + 2\,u_2\,u_3\,v_{41} - 4\,u_2\,u_4\,v_{31} + 2\,u_3\,u_4\,v_{21} + 2\,c_5, -4\,u_1\,u_2\,v_{43} + 2\,u_1\,u_3\,v_{42} + 2\,u_1\,u_4\,v_{32} + 2\,u_2\,u_3\,v_{41} + 2\,u_2\,u_4\,v_{31} - 4\,u_3\,u_4\,v_{21} + 2\,c_9, -2\,u_1^2\,v_{32} + 2\,u_1\,u_2\,v_{31} + 2\,u_1\,u_3\,v_{21} - 2\,u_2\,u_3\,v_{11} + 2\,c_1, -2\,u_1^2\,v_{42} + 2\,u_1\,u_2\,v_{41} + 2\,u_1\,u_4\,v_{21} - 2\,u_2\,u_4\,v_{11} + 2\,c_2, -u_3^2\,v_{44} + 2\,u_3\,u_4\,v_{43} - u_4^2\,v_{33} + c_{20}\right] \tag{4.6}$$

It only remains to eleminate the u's and the v's.
WARNING : the computation below requires ~ 10GB of RAM.

```
> select(has, eqsChowConic, c[12]);
```

$$[\,] \tag{4.7}$$

```
> # gbChowConic := fgb_gbasis_elim(eqsChowConic, 0, [op(indets
  (eqsChowConic) minus {op(cvars)})], cvars0, {"verb" = 3, "index"
  = 10^8}):
```

Consistency check

```
> map(normal, subs(solve(eqsChowConic, cvars)[1], gbChowConic));
```

$$[0, 0] \tag{4.8}$$

[Section 5] Ideal of the Hurwitz forms of quadrics in P^3

A line in P^3, given by Stiefel coordinates is tangent to the quadric defined by f2 if and only if the following discriminant vanishes:

```
> disc := collect(discrim(f2, t), stiefelVars, 'distributed');
```

$$disc := \left(-4 v_{11} v_{22} + 4 v_{21}^2\right) a_{11}^2 a_{22}^2 + \left(-4 v_{11} v_{33} + 4 v_{31}^2\right) a_{11}^2 a_{23}^2 + \left(-4 v_{11} v_{44} + 4 v_{41}^2\right) a_{11}^2 a_{24}^2 + \left(-4 v_{11} v_{22} + 4 v_{21}^2\right) a_{12}^2 a_{21}^2 + \left(-4 v_{22} v_{33} + 4 v_{32}^2\right) a_{12}^2 a_{23}^2 + \left(-4 v_{22} v_{44} + 4 v_{42}^2\right) a_{12}^2 a_{24}^2 + \left(-4 v_{11} v_{33} + 4 v_{31}^2\right) a_{13}^2 a_{21}^2 + \left(-4 v_{22} v_{33} + 4 v_{32}^2\right) a_{13}^2 a_{22}^2 + \left(-4 v_{33} v_{44} + 4 v_{43}^2\right) a_{13}^2 a_{24}^2 + \left(-4 v_{11} v_{44} + 4 v_{41}^2\right) a_{14}^2 a_{21}^2 + \left(-4 v_{22} v_{44} + 4 v_{42}^2\right) a_{14}^2 a_{22}^2 + \left(-4 v_{33} v_{44} + 4 v_{43}^2\right) a_{14}^2 a_{23}^2 + \left(8 v_{22} v_{43} - 8 v_{32} v_{42}\right) a_{12} a_{14} a_{22} a_{23} + \left(8 v_{22} v_{44} - 8 v_{42}^2\right) a_{12} a_{14} a_{22} a_{24} + \left(8 v_{32} v_{44} - 8 v_{42} v_{43}\right) a_{12} a_{14} a_{23} a_{24} + \left(-16 v_{21} v_{43} + 8 v_{31} v_{42} + 8 v_{32} v_{41}\right) a_{13} a_{14} a_{21} a_{22} + \left(-8 v_{31} v_{43} + 8 v_{33} v_{41}\right) a_{13} a_{14} a_{21} a_{23} + \left(8 v_{31} v_{44} - 8 v_{41} v_{43}\right) a_{13} a_{14} a_{21} a_{24} + \left(-8 v_{32} v_{43} + 8 v_{33} v_{42}\right) a_{13} a_{14} a_{22} a_{23} + \left(8 v_{32} v_{44} - 8 v_{42} v_{43}\right) a_{13} a_{14} a_{22} a_{24} + \left(8 v_{33} v_{44} - 8 v_{43}^2\right) a_{13} a_{14} a_{23} a_{24} + \left(8 v_{11} v_{22} - 8 v_{21}^2\right) a_{11} a_{12} a_{21} a_{22} + \left(8 v_{11} v_{32} - 8 v_{21} v_{31}\right) a_{11} a_{12} a_{21} a_{23} + \left(8 v_{11} v_{42} - 8 v_{21} v_{41}\right) a_{11} a_{12} a_{21} a_{24} + \left(-8 v_{21} v_{32} + 8 v_{22} v_{31}\right) a_{11} a_{12} a_{22} a_{23} + \left(-8 v_{21} v_{42} + 8 v_{22} v_{41}\right) a_{11} a_{12} a_{22} a_{24} + \left(-16 v_{21} v_{43} + 8 v_{31} v_{42} + 8 v_{32} v_{41}\right) a_{11} a_{12} a_{23} a_{24} + \left(8 v_{11} v_{32} - 8 v_{21} v_{31}\right) a_{11} a_{13} a_{21} a_{22} + \left(8 v_{11} v_{33} - 8 v_{31}^2\right) a_{11} a_{13} a_{21} a_{23} + \left(8 v_{11} v_{43} - 8 v_{31} v_{41}\right) a_{11} a_{13} a_{21} a_{24} + \left(8 v_{21} v_{33} - 8 v_{31} v_{32}\right) a_{11} a_{13} a_{22} a_{23} + \left(8 v_{21} v_{43} - 16 v_{31} v_{42} + 8 v_{32} v_{41}\right) a_{11} a_{13} a_{22} a_{24} + \left(-8 v_{31} v_{43} + 8 v_{33} v_{41}\right) a_{11} a_{13} a_{23} a_{24} + \left(8 v_{11} v_{42} - 8 v_{21} v_{41}\right) a_{11} a_{14} a_{21} a_{22} + \left(8 v_{11} v_{43} - 8 v_{31} v_{41}\right) a_{11} a_{14} a_{21} a_{23} + \left(8 v_{11} v_{44} - 8 v_{41}^2\right) a_{11} a_{14} a_{21} a_{24} + \left(8 v_{21} v_{43} + 8 v_{31} v_{42} - 16 v_{32} v_{41}\right) a_{11} a_{14} a_{22} a_{23} + \left(8 v_{21} v_{44} - 8 v_{41} v_{42}\right) a_{11} a_{14} a_{22} a_{24} + \left(8 v_{31} v_{44} - 8 v_{41} v_{43}\right) a_{11} a_{14} a_{23} a_{24} + \left(-8 v_{21} v_{32} + 8 v_{22} v_{31}\right) a_{12} a_{13} a_{21} a_{22} + \left(8 v_{21} v_{33} - 8 v_{31} v_{32}\right) a_{12} a_{13} a_{21} a_{23} + \left(8 v_{21} v_{43} + 8 v_{31} v_{42} - 16 v_{32} v_{41}\right) a_{12} a_{13} a_{21} a_{24} + \left(8 v_{22} v_{33} - 8 v_{32}^2\right) a_{12} a_{13} a_{22} a_{23} + \left(8 v_{22} v_{43} - 8 v_{32} v_{42}\right) a_{12} a_{13} a_{22} a_{24} + \left(-8 v_{32} v_{43} + 8 v_{33} v_{42}\right) a_{12} a_{13} a_{23} a_{24} + \left(-8 v_{21} v_{42} + 8 v_{22} v_{41}\right) a_{12} a_{14} a_{21} a_{22} + \left(8 v_{21} v_{43} - 16 v_{31} v_{42} + 8 v_{32} v_{41}\right) a_{12} a_{14} a_{21} a_{23} + \left(8 v_{21} v_{44} - 8 v_{41} v_{42}\right) a_{12} a_{14} a_{21} a_{24} + \left(-8 v_{11} v_{32} + 8 v_{21} v_{31}\right) a_{11}^2 a_{22} a_{23} + \left(-8 v_{11} v_{42} + 8 v_{21} v_{41}\right) a_{11}^2 a_{22} a_{24} + \left(-8 v_{11} v_{43} + 8 v_{31} v_{41}\right) a_{11}^2 a_{23} a_{24} + \left(-8 v_{21} v_{33}\right. \tag{5.1}$$

$$+ 8 v_{31} v_{32}) a_{11} a_{12} a_{23}^2 + (-8 v_{21} v_{44} + 8 v_{41} v_{42}) a_{11} a_{12} a_{24}^2 + (8 v_{21} v_{32} - 8 v_{22} v_{31}) a_{11} a_{13} a_{22}^2 + (-8 v_{31} v_{44} + 8 v_{41} v_{43}) a_{11} a_{13} a_{24}^2 + (8 v_{21} v_{42} - 8 v_{22} v_{41}) a_{11} a_{14} a_{22}^2 + (8 v_{31} v_{43} - 8 v_{33} v_{41}) a_{11} a_{14} a_{23}^2 + (8 v_{21} v_{32} - 8 v_{22} v_{31}) a_{12}^2 a_{21} a_{23} + (8 v_{21} v_{42} - 8 v_{22} v_{41}) a_{12}^2 a_{21} a_{24} + (-8 v_{22} v_{43} + 8 v_{32} v_{42}) a_{12}^2 a_{23} a_{24} + (-8 v_{11} v_{32} + 8 v_{21} v_{31}) a_{12} a_{13} a_{21}^2 + (-8 v_{32} v_{44} + 8 v_{42} v_{43}) a_{12} a_{13} a_{24}^2 + (-8 v_{11} v_{42} + 8 v_{21} v_{41}) a_{12} a_{14} a_{21}^2 + (8 v_{32} v_{43} - 8 v_{33} v_{42}) a_{12} a_{14} a_{23}^2 + (-8 v_{21} v_{33} + 8 v_{31} v_{32}) a_{13}^2 a_{21} a_{22} + (8 v_{31} v_{43} - 8 v_{33} v_{41}) a_{13}^2 a_{21} a_{24} + (8 v_{32} v_{43} - 8 v_{33} v_{42}) a_{13}^2 a_{22} a_{24} + (-8 v_{11} v_{43} + 8 v_{31} v_{41}) a_{13} a_{14} a_{21}^2 + (-8 v_{22} v_{43} + 8 v_{32} v_{42}) a_{13} a_{14} a_{22}^2 + (-8 v_{21} v_{44} + 8 v_{41} v_{42}) a_{14}^2 a_{21} a_{22} + (-8 v_{31} v_{44} + 8 v_{41} v_{43}) a_{14}^2 a_{21} a_{23} + (-8 v_{32} v_{44} + 8 v_{42} v_{43}) a_{14}^2 a_{22} a_{23}$$

This can be expressed in terms of the Plucker coordinates.

```
> pluckerDisc := fgb_gbasis_elim([disc, seq(pluckerVars[i]-subs
  (pluckerCo,pluckerVars[i]), i=1..6)], 0, stiefelVars, [op(indets
  (disc) minus {op(stiefelVars)}), op(pluckerVars)])[1];
```

$$pluckerDisc := -p_{0,1}^2 v_{11} v_{22} + p_{0,1}^2 v_{21}^2 - 2 p_{0,1} p_{0,2} v_{11} v_{32} + 2 p_{0,1} p_{0,2} v_{21} v_{31} - 2 p_{0,1} p_{0,3} v_{11} v_{42} + 2 p_{0,1} p_{0,3} v_{21} v_{41} - 2 p_{0,1} p_{1,2} v_{21} v_{32} + 2 p_{0,1} p_{1,2} v_{22} v_{31} - 2 p_{0,1} p_{1,3} v_{21} v_{42} + 2 p_{0,1} p_{1,3} v_{22} v_{41} + 2 p_{0,1} p_{2,3} v_{21} v_{43} - 4 p_{0,1} p_{2,3} v_{31} v_{42} + 2 p_{0,1} p_{2,3} v_{32} v_{41} - p_{0,2}^2 v_{11} v_{33} + p_{0,2}^2 v_{31}^2 - 2 p_{0,2} p_{0,3} v_{11} v_{43} + 2 p_{0,2} p_{0,3} v_{31} v_{41} - 2 p_{0,2} p_{1,2} v_{21} v_{33} + 2 p_{0,2} p_{1,2} v_{31} v_{32} - 4 p_{0,2} p_{1,3} v_{21} v_{43} + 2 p_{0,2} p_{1,3} v_{31} v_{42} + 2 p_{0,2} p_{1,3} v_{32} v_{41} - 2 p_{0,2} p_{2,3} v_{31} v_{43} + 2 p_{0,2} p_{2,3} v_{33} v_{41} - p_{0,3}^2 v_{11} v_{44} + p_{0,3}^2 v_{41}^2 - 2 p_{0,3} p_{1,3} v_{21} v_{44} + 2 p_{0,3} p_{1,3} v_{41} v_{42} - 2 p_{0,3} p_{2,3} v_{31} v_{44} + 2 p_{0,3} p_{2,3} v_{41} v_{43} - p_{1,2}^2 v_{22} v_{33} + p_{1,2}^2 v_{32}^2 - 2 p_{1,2} p_{1,3} v_{22} v_{43} + 2 p_{1,2} p_{1,3} v_{32} v_{42} - 2 p_{1,2} p_{2,3} v_{32} v_{43} + 2 p_{1,2} p_{2,3} v_{33} v_{42} - p_{1,3}^2 v_{22} v_{44} + p_{1,3}^2 v_{42}^2 - 2 p_{1,3} p_{2,3} v_{32} v_{44} + 2 p_{1,3} p_{2,3} v_{42} v_{43} - p_{2,3}^2 v_{33} v_{44} + p_{2,3}^2 v_{43}^2 \quad (5.2)$$

As above, we look for the conditions that gQ equals pluckerDisc modulo the Plucker relation.

```
> Groebner[NormalForm](gQ - pluckerDisc, [pluckerRel], tdeg(op
  (pluckerVars)));
```

$$(2 v_{21} v_{42} - 2 v_{22} v_{41} + 2 c_4) p_{0,1} p_{1,3} + (2 v_{21} v_{44} - 2 v_{41} v_{42} + 2 c_{13}) p_{0,3} p_{1,3} + (2 v_{22} v_{43} - 2 v_{32} v_{42} + 2 c_{16}) p_{1,2} p_{1,3} + (-2 v_{21} v_{43} + 4 v_{31} v_{42} - 2 v_{32} v_{41} \quad (5.3)$$

$$+ 2\, c_{5}\big) p_{0,1}\, p_{2,3} + \big(v_{22}\, v_{44} - v_{42}^{2} + c_{18}\big) p_{1,3}^{2} + \big(4\, v_{21}\, v_{43} - 2\, v_{31}\, v_{42} - 2\, v_{32}\, v_{41} + 2\, c_{9}\big) p_{0,2}\, p_{1,3} + \big(v_{11}\, v_{22} - v_{21}^{2} + c_{0}\big) p_{0,1}^{2} + \big(v_{11}\, v_{33} - v_{31}^{2} + c_{6}\big) p_{0,2}^{2} + \big(v_{11}\, v_{44} - v_{41}^{2} + c_{11}\big) p_{0,3}^{2} + \big(v_{22}\, v_{33} - v_{32}^{2} + c_{15}\big) p_{1,2}^{2} + \big(2\, v_{11}\, v_{32} - 2\, v_{21}\, v_{31} + 2\, c_{1}\big) p_{0,1}\, p_{0,2} + \big(2\, v_{11}\, v_{42} - 2\, v_{21}\, v_{41} + 2\, c_{2}\big) p_{0,1}\, p_{0,3} + \big(2\, v_{11}\, v_{43} - 2\, v_{31}\, v_{41} + 2\, c_{7}\big) p_{0,2}\, p_{0,3} + \big(2\, v_{21}\, v_{32} - 2\, v_{22}\, v_{31} + 2\, c_{3}\big) p_{0,1}\, p_{1,2} + \big(2\, v_{21}\, v_{33} - 2\, v_{31}\, v_{32} + 2\, c_{8}\big) p_{0,2}\, p_{1,2} + \big(v_{33}\, v_{44} - v_{43}^{2} + c_{20}\big) p_{2,3}^{2} + \big(2\, v_{31}\, v_{43} - 2\, v_{33}\, v_{41} + 2\, c_{10}\big) p_{0,2}\, p_{2,3} + \big(2\, v_{31}\, v_{44} - 2\, v_{41}\, v_{43} + 2\, c_{14}\big) p_{0,3}\, p_{2,3} + \big(2\, v_{32}\, v_{43} - 2\, v_{33}\, v_{42} + 2\, c_{17}\big) p_{1,2}\, p_{2,3} + \big(2\, v_{32}\, v_{44} - 2\, v_{42}\, v_{43} + 2\, c_{19}\big) p_{1,3}\, p_{2,3}$$

```
> eqsHurwitz := [coeffs(%, pluckerVars)];
```

$$eqsHurwitz := \big[2\, v_{21}\, v_{42} - 2\, v_{22}\, v_{41} + 2\, c_{4}, 2\, v_{21}\, v_{44} - 2\, v_{41}\, v_{42} + 2\, c_{13}, 2\, v_{22}\, v_{43} - 2\, v_{32}\, v_{42} + 2\, c_{16}, -2\, v_{21}\, v_{43} + 4\, v_{31}\, v_{42} - 2\, v_{32}\, v_{41} + 2\, c_{5}, v_{22}\, v_{44} - v_{42}^{2} + c_{18}, 4\, v_{21}\, v_{43} - 2\, v_{31}\, v_{42} - 2\, v_{32}\, v_{41} + 2\, c_{9}, v_{11}\, v_{22} - v_{21}^{2} + c_{0}, v_{11}\, v_{33} - v_{31}^{2} + c_{6}, v_{11}\, v_{44} - v_{41}^{2} + c_{11}, v_{22}\, v_{33} - v_{32}^{2} + c_{15}, 2\, v_{11}\, v_{32} - 2\, v_{21}\, v_{31} + 2\, c_{1}, 2\, v_{11}\, v_{42} - 2\, v_{21}\, v_{41} + 2\, c_{2}, 2\, v_{11}\, v_{43} - 2\, v_{31}\, v_{41} + 2\, c_{7}, 2\, v_{21}\, v_{32} - 2\, v_{22}\, v_{31} + 2\, c_{3}, 2\, v_{21}\, v_{33} - 2\, v_{31}\, v_{32} + 2\, c_{8}, v_{33}\, v_{44} - v_{43}^{2} + c_{20}, 2\, v_{31}\, v_{43} - 2\, v_{33}\, v_{41} + 2\, c_{10}, 2\, v_{31}\, v_{44} - 2\, v_{41}\, v_{43} + 2\, c_{14}, 2\, v_{32}\, v_{43} - 2\, v_{33}\, v_{42} + 2\, c_{17}, 2\, v_{32}\, v_{44} - 2\, v_{42}\, v_{43} + 2\, c_{19}\big] \tag{5.4}$$

WARNING : the computation below requires some RAM.

```
> # gbHurwitz := fgb_gbasis_elim(eqsHurwitz, 0, [op(indets
  (eqsHurwitz) minus {op(cvars)})], cvars0, {"verb" = 3, "index" =
  10^8}):
```

[Section 5bis] Ideal of the Hurwitz forms of quadrics in P^3 (alternative method)

See B. Sturmfels: The Hurwitz form of a projective variety, Example 2.2, for more details about the method.

```
> sym4 := Matrix(4, (i,j) -> d[i+10*j], shape = symmetric);
```

$$sym4 := \begin{bmatrix} d_{11} & d_{21} & d_{31} & d_{41} \\ d_{21} & d_{22} & d_{32} & d_{42} \\ d_{31} & d_{32} & d_{33} & d_{43} \\ d_{41} & d_{42} & d_{43} & d_{44} \end{bmatrix} \tag{6.1}$$

```
> choose24 := combinat[choose]([1,2,3,4], 2);
```

(6.2)

$$choose24 := [[1, 2], [1, 3], [1, 4], [2, 3], [2, 4], [3, 4]] \quad (6.2)$$

```
> extSym4 := Matrix(6, (i,j) -> Determinant(sym4[choose24[i],
  choose24[j]]));
```

$$extSym4 := [[d_{11} d_{22} - d_{21}^2, d_{11} d_{32} - d_{21} d_{31}, d_{11} d_{42} - d_{21} d_{41}, d_{21} d_{32} - d_{22} d_{31}, d_{21} d_{42} - d_{22} d_{41}, d_{31} d_{42} - d_{32} d_{41}],$$
$$[d_{11} d_{32} - d_{21} d_{31}, d_{11} d_{33} - d_{31}^2, d_{11} d_{43} - d_{31} d_{41}, d_{21} d_{33} - d_{31} d_{32}, d_{21} d_{43} - d_{32} d_{41}, d_{31} d_{43} - d_{33} d_{41}],$$
$$[d_{11} d_{42} - d_{21} d_{41}, d_{11} d_{43} - d_{31} d_{41}, d_{11} d_{44} - d_{41}^2, d_{21} d_{43} - d_{31} d_{42}, d_{21} d_{44} - d_{41} d_{42}, d_{31} d_{44} - d_{41} d_{43}],$$
$$[d_{21} d_{32} - d_{22} d_{31}, d_{21} d_{33} - d_{31} d_{32}, d_{21} d_{43} - d_{31} d_{42}, d_{22} d_{33} - d_{32}^2, d_{22} d_{43} - d_{32} d_{42}, d_{32} d_{43} - d_{33} d_{42}],$$
$$[d_{21} d_{42} - d_{22} d_{41}, d_{21} d_{43} - d_{32} d_{41}, d_{21} d_{44} - d_{41} d_{42}, d_{22} d_{43} - d_{32} d_{42}, d_{22} d_{44} - d_{42}^2, d_{32} d_{44} - d_{42} d_{43}],$$
$$[d_{31} d_{42} - d_{32} d_{41}, d_{31} d_{43} - d_{33} d_{41}, d_{31} d_{44} - d_{41} d_{43}, d_{32} d_{43} - d_{33} d_{42}, d_{32} d_{44} - d_{42} d_{43}, d_{33} d_{44} - d_{43}^2]] \quad (6.3)$$

```
> gHurwitz := collect(Vector[row](pluckerVars).extSym4.Vector
  (pluckerVars), pluckerVars, distributed);
```

$$gHurwitz := (d_{11} d_{22} - d_{21}^2) p_{0,1}^2 + (2 d_{21} d_{42} - 2 d_{22} d_{41}) p_{1,3} p_{0,1} + (2 d_{31} d_{42} - 2 d_{32} d_{41}) p_{2,3} p_{0,1} + (2 d_{11} d_{32} - 2 d_{21} d_{31}) p_{0,2} p_{0,1} + (2 d_{11} d_{42} - 2 d_{21} d_{41}) p_{0,3} p_{0,1} + (2 d_{21} d_{32} - 2 d_{22} d_{31}) p_{1,2} p_{0,1} + (d_{11} d_{33} - d_{31}^2) p_{0,2}^2 + (2 d_{21} d_{43} - 2 d_{32} d_{41}) p_{1,3} p_{0,2} + (2 d_{31} d_{43} - 2 d_{33} d_{41}) p_{2,3} p_{0,2} + (2 d_{11} d_{43} - 2 d_{31} d_{41}) p_{0,3} p_{0,2} + (2 d_{21} d_{33} - 2 d_{31} d_{32}) p_{1,2} p_{0,2} + (d_{11} d_{44} - d_{41}^2) p_{0,3}^2 + (2 d_{21} d_{44} - 2 d_{41} d_{42}) p_{1,3} p_{0,3} + (2 d_{31} d_{44} - 2 d_{41} d_{43}) p_{2,3} p_{0,3} + (2 d_{21} d_{43} - 2 d_{31} d_{42}) p_{1,2} p_{0,3} + (d_{22} d_{33} - d_{32}^2) p_{1,2}^2 + (2 d_{22} d_{43} - 2 d_{32} d_{42}) p_{1,3} p_{1,2} + (2 d_{32} d_{43} - 2 d_{33} d_{42}) p_{2,3} p_{1,2} + (d_{22} d_{44} - d_{42}^2) p_{1,3}^2 + (2 d_{32} d_{44} - 2 d_{42} d_{43}) p_{2,3} p_{1,3} + p_{2,3}^2 (d_{33} d_{44} - d_{43}^2) \quad (6.4)$$

```
> Groebner[Reduce](gQ - gHurwitz, [pluckerRel], tdeg(op
  (pluckerVars)));
```

$$(-d_{11} d_{33} + d_{31}^2 + c_6) p_{0,2}^2 + (-d_{11} d_{44} + d_{41}^2 + c_{11}) p_{0,3}^2 + (-d_{22} d_{33} + d_{32}^2 + c_{15}) p_{1,2}^2 + (-d_{22} d_{44} + d_{42}^2 + c_{18}) p_{1,3}^2 + (-d_{11} d_{22} + d_{21}^2 + c_0) p_{0,1}^2 + (-2 d_{11} d_{43} + 2 d_{31} d_{41} + 2 c_7) p_{0,3} p_{0,2} + (-2 d_{32} d_{43} + 2 d_{33} d_{42} + 2 c_{17}) p_{2,3} p_{1,2} + (\quad (6.5)$$

$$-2\, d_{21}\, d_{33} + 2\, d_{31}\, d_{32} + 2\, c_8\big)\, p_{1,2}\, p_{0,2} + \big(-2\, d_{21}\, d_{42} + 2\, d_{22}\, d_{41} + 2\, c_4\big)\, p_{1,3}\, p_{0,1} + \big(-2\, d_{31}\, d_{44} + 2\, d_{41}\, d_{43} + 2\, c_{14}\big)\, p_{2,3}\, p_{0,3} + \big(-4\, d_{21}\, d_{43} + 2\, d_{31}\, d_{42} + 2\, d_{32}\, d_{41} + 2\, c_9\big)\, p_{1,3}\, p_{0,2} + \big(-2\, d_{32}\, d_{44} + 2\, d_{42}\, d_{43} + 2\, c_{19}\big)\, p_{2,3}\, p_{1,3} + \big(-2\, d_{31}\, d_{43} + 2\, d_{33}\, d_{41} + 2\, c_{10}\big)\, p_{2,3}\, p_{0,2} + \big(-2\, d_{22}\, d_{43} + 2\, d_{32}\, d_{42} + 2\, c_{16}\big)\, p_{1,3}\, p_{1,2} + \big(-2\, d_{21}\, d_{32} + 2\, d_{22}\, d_{31} + 2\, c_3\big)\, p_{1,2}\, p_{0,1} + \big(-2\, d_{21}\, d_{44} + 2\, d_{41}\, d_{42} + 2\, c_{13}\big)\, p_{1,3}\, p_{0,3} + \big(-2\, d_{11}\, d_{42} + 2\, d_{21}\, d_{41} + 2\, c_2\big)\, p_{0,3}\, p_{0,1} + \big(-2\, d_{11}\, d_{32} + 2\, d_{21}\, d_{31} + 2\, c_1\big)\, p_{0,2}\, p_{0,1} + \big(2\, d_{21}\, d_{43} - 4\, d_{31}\, d_{42} + 2\, d_{32}\, d_{41} + 2\, c_5\big)\, p_{2,3}\, p_{0,1} + \big(-d_{33}\, d_{44} + d_{43}^2 + c_{20}\big)\, p_{2,3}^2$$

```
> eqsHurwitzAlt := [coeffs(%, pluckerVars)];
```

$$eqsHurwitzAlt := \big[-d_{11}\, d_{33} + d_{31}^2 + c_6,\ -d_{11}\, d_{44} + d_{41}^2 + c_{11},\ -d_{22}\, d_{33} + d_{32}^2 + c_{15},\ -d_{22}\, d_{44} + d_{42}^2 + c_{18},\ -d_{11}\, d_{22} + d_{21}^2 + c_0,\ -2\, d_{11}\, d_{43} + 2\, d_{31}\, d_{41} + 2\, c_7,\ -2\, d_{32}\, d_{43} + 2\, d_{33}\, d_{42} + 2\, c_{17},\ -2\, d_{21}\, d_{33} + 2\, d_{31}\, d_{32} + 2\, c_8,\ -2\, d_{21}\, d_{42} + 2\, d_{22}\, d_{41} + 2\, c_4,\ -2\, d_{31}\, d_{44} + 2\, d_{41}\, d_{43} + 2\, c_{14},\ -4\, d_{21}\, d_{43} + 2\, d_{31}\, d_{42} + 2\, d_{32}\, d_{41} + 2\, c_9,\ -2\, d_{32}\, d_{44} + 2\, d_{42}\, d_{43} + 2\, c_{19},\ -2\, d_{31}\, d_{43} + 2\, d_{33}\, d_{41} + 2\, c_{10},\ -2\, d_{22}\, d_{43} + 2\, d_{32}\, d_{42} + 2\, c_{16},\ -2\, d_{21}\, d_{32} + 2\, d_{22}\, d_{31} + 2\, c_3,\ -2\, d_{21}\, d_{44} + 2\, d_{41}\, d_{42} + 2\, c_{13},\ -2\, d_{11}\, d_{42} + 2\, d_{21}\, d_{41} + 2\, c_2,\ -2\, d_{11}\, d_{32} + 2\, d_{21}\, d_{31} + 2\, c_1,\ 2\, d_{21}\, d_{43} - 4\, d_{31}\, d_{42} + 2\, d_{32}\, d_{41} + 2\, c_5,\ -d_{33}\, d_{44} + d_{43}^2 + c_{20}\big] \quad \textbf{(6.6)}$$

```
> nops(select(has, eqsHurwitzAlt, c[12]));
```

$$0 \quad \textbf{(6.7)}$$

It only remains to eleminate the d's
WARNING : the computation below requires some RAM.

```
> # gbHurwitzAlt := fgb_gbasis_elim(eqsHurwitzAlt, 0, [op(indets
  (eqsHurwitzAlt) minus {op(cvars)})], cvars0, {"verb" = 3, "index"
  = 10^8}):
> # evalb(gbHurwitz = gbHurwitzAlt);
```

[Section 6] Ideal of Chow forms that are squares

```
> Groebner[Reduce](gQ - add(e[i]*pluckerVars[i], i=1..6)^2,
  [pluckerRel], tdeg(op(pluckerVars)));
```

$$\big(-e_2^2 + c_6\big)\, p_{0,2}^2 + \big(-e_3^2 + c_{11}\big)\, p_{0,3}^2 + \big(-e_4^2 + c_{15}\big)\, p_{1,2}^2 + \big(-e_5^2 + c_{18}\big)\, p_{1,3}^2 + \big(-e_1^2 + c_0\big)\, p_{0,1}^2 + \big(-2\, e_2\, e_3 + 2\, c_7\big)\, p_{0,3}\, p_{0,2} + \big(-2\, e_4\, e_6 + 2\, c_{17}\big)\, p_{2,3}\, p_{1,2} + \big(-2\, e_2\, e_4 + 2\, c_8\big)\, p_{1,2}\, p_{0,2} + \big(-2\, e_1\, e_5 + 2\, c_4\big)\, p_{1,3}\, p_{0,1} + \big(-2\, e_3\, e_6 \quad \textbf{(7.1)}$$

$$+2c_{14})p_{2,3}p_{0,3} + (-2e_2e_5 - 2e_3e_4 + 2c_9)p_{1,3}p_{0,2} + (-2e_5e_6 + 2c_{19})p_{2,3}p_{1,3} + (-2e_2e_6 + 2c_{10})p_{2,3}p_{0,2} + (-2e_4e_5 + 2c_{16})p_{1,3}p_{1,2} + (-2e_1e_4 + 2c_3)p_{1,2}p_{0,1} + (-2e_3e_5 + 2c_{13})p_{1,3}p_{0,3} + (-2e_1e_3 + 2c_2)p_{0,3}p_{0,1} + (-2e_1e_2 + 2c_1)p_{0,2}p_{0,1} + (-2e_1e_6 + 2e_3e_4 + 2c_5)p_{2,3}p_{0,1} + (-e_6^2 + c_{20})p_{2,3}^2$$

```
> eqsSquare := [coeffs(%, pluckerVars)];
```

$$eqsSquare := [-e_2^2 + c_6, -e_3^2 + c_{11}, -e_4^2 + c_{15}, -e_5^2 + c_{18}, -e_1^2 + c_0, -2e_2e_3 + 2c_7, -2e_4e_6 + 2c_{17}, -2e_2e_4 + 2c_8, -2e_1e_5 + 2c_4, -2e_3e_6 + 2c_{14}, -2e_2e_5 - 2e_3e_4 + 2c_9, -2e_5e_6 + 2c_{19}, -2e_2e_6 + 2c_{10}, -2e_4e_5 + 2c_{16}, -2e_1e_4 + 2c_3, -2e_3e_5 + 2c_{13}, -2e_1e_3 + 2c_2, -2e_1e_2 + 2c_1, -2e_1e_6 + 2e_3e_4 + 2c_5, -e_6^2 + c_{20}] \tag{7.2}$$

```
> select(has, eqsSquare, c[12]);
```

$$[\,] \tag{7.3}$$

```
> [op(indets(eqsSquare) minus {op(cvars)})], cvars0;
```

$$[e_1, e_2, e_3, e_4, e_5, e_6], [c_0, c_1, c_2, c_3, c_4, c_5, c_6, c_7, c_8, c_9, c_{10}, c_{11}, c_{13}, c_{14}, c_{15}, c_{16}, c_{17}, c_{18}, c_{19}, c_{20}] \tag{7.4}$$

It only remains to eleminate the e's.

```
> gbSquare := fgb_gbasis_elim(eqsSquare, 0, %, {"verb" = 3}):
```

[Section 7] Some inclusions

```
> map(Groebner[Reduce], gensCoiso, gbChowConic, tdeg(op(cvars0)));
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0] **(8.1)**

```
> map(Groebner[Reduce], gensCoiso, gbChowLines, tdeg(op(cvars0)));
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, **(8.2)**

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]

```
> map(Groebner[Reduce], gensCoiso, gbHurwitz, tdeg(op(cvars0)));
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, **(8.3)**
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]

```
> map(Groebner[Reduce], gensCoiso, gbSquare, tdeg(op(cvars0)));
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, **(8.4)**
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]

This is Proposition 2 in “Computing the Chow variety of quadratic space curves”

```
> map(Groebner[Reduce], gbHurwitz, gbChowConic, tdeg(op(cvars0)));
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, **(8.5)**
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]

[Section 8] Primary decomposition of the coisotropic ideal

We now check that the coisotropic ideal is the intersection of the ideals corresponding to Chow forms of conics, Chow form of skew lines and Hurwitz form of quadrics.

```
> eqsInter := [1-u[1]-u[2]-u[3], seq(u[1]*p, p in gbSquare), seq(u
  [2]*p, p in gbHurwitz), seq(u[3]*p, p in gbChowLines)]:
```

This computes Groebner basis of the intersection.

```
> gbInter := fgb_gbasis_elim(eqsInter, 0, [u[1], u[2], u[3]],
  cvars0):
```

This is Proposition 1 in “Computing the Chow variety of quadratic space curves”

```
> evalb(gbInter = gbCoiso);
```

$$true \tag{9.1}$$

[Section 9] Ideal of the Chow variety

The Chow variety G(2,2,4) is formed by the Chow forms of all plane cubics and all pairs of skew lines.

```
> eqsChowForm := [seq((1-t)*p, p in gbChowLines), seq(t*p, p in
  gbChowConic)]:
> gbChowForm := fgb_gbasis_elim(eqsChowForm, 0, [t], cvars0): nops
  (%);
```

$$348 \tag{10.1}$$

[Section 10] The integrability ideal

A few functions to deal with differential forms.

```
> read "Df.mpl": with(Df[MM]):
```

Here, we deal with affine the following affine chart of the Grassmaniann.

```
> affineCo := [a[11]=1,a[21]=0,a[12]=0,a[22]=1];
```

$$affineCo := [a_{11} = 1, a_{21} = 0, a_{12} = 0, a_{22} = 1] \tag{11.1}$$

```
> affineVars := remove(type, subs(affineCo, stiefelVars), integer);
```

$$affineVars := [a_{13}, a_{14}, a_{23}, a_{24}] \tag{11.2}$$

```
> pluckerToAffine := subs(affineCo, pluckerCo);
```

$$pluckerToAffine := [p_{0,1} = 1, p_{0,2} = a_{23}, p_{0,3} = a_{24}, p_{1,2} = -a_{13}, p_{1,3} = -a_{14}, p_{2,3} = a_{13}\,a_{24} - a_{14}\,a_{23}] \tag{11.3}$$

```
> alg := map(v->d[v], convert(affineVars, set));
```

$$alg := \{d_{a_{13}}, d_{a_{14}}, d_{a_{23}}, d_{a_{24}}\} \tag{11.4}$$

Note that we use dual coordinates.

```
> affgQ := subs(pluckerDual, pluckerToAffine, gQ);
```

$$affgQ := c_0\left(a_{13}a_{24}-a_{14}a_{23}\right)^2-2c_4a_{23}\left(a_{13}a_{24}-a_{14}a_{23}\right)+\left(2c_5+2c_{12}\right)\left(a_{13}a_{24}-a_{14}a_{23}\right)+2c_1a_{14}\left(a_{13}a_{24}-a_{14}a_{23}\right)-2c_2a_{13}\left(a_{13}a_{24}-a_{14}a_{23}\right)+2c_3a_{24}\left(a_{13}a_{24}-a_{14}a_{23}\right)+c_6a_{14}^2-\left(2c_9-2c_{12}\right)a_{14}a_{23}+2c_{10}a_{14}-2c_7a_{13}a_{14}+2c_8a_{24}a_{14}+c_{11}a_{13}^2-2c_{12}a_{13}a_{24}+2c_{13}a_{23}a_{13}-2c_{14}a_{13}+c_{15}a_{24}^2-2c_{16}a_{23}a_{24}+2c_{17}a_{24}+c_{18}a_{23}^2-2c_{19}a_{23}+c_{20} \quad \textbf{(11.5)}$$

```
> affineMat := subs(affineCo, stiefelMat);
```

$$affineMat := \begin{bmatrix} 1 & 0 \\ 0 & 1 \\ a_{13} & a_{23} \\ a_{14} & a_{24} \end{bmatrix} \quad \textbf{(11.6)}$$

```
> alphas := [seq(seq(add(diff(affgQ,affineMat[l+2,i])*d[affineMat
  [l+2,j]],l=1..2),j=1..2),i=1..2)];
```

$$alphas := \left[\left(2c_0\left(a_{13}a_{24}-a_{14}a_{23}\right)a_{24}-2c_4a_{23}a_{24}+\left(2c_5+2c_{12}\right)a_{24}+2c_1a_{14}a_{24}-2c_2\left(a_{13}a_{24}-a_{14}a_{23}\right)-2c_2a_{13}a_{24}+2c_3a_{24}^2-2c_7a_{14}+2c_{11}a_{13}-2c_{12}a_{24}+2c_{13}a_{23}-2c_{14}\right)d_{a_{13}}+\left(-2c_0\left(a_{13}a_{24}-a_{14}a_{23}\right)a_{23}+2c_4a_{23}^2-\left(2c_5+2c_{12}\right)a_{23}+2c_1\left(a_{13}a_{24}-a_{14}a_{23}\right)-2c_1a_{14}a_{23}+2c_2a_{13}a_{23}-2c_3a_{24}a_{23}+2c_6a_{14}-\left(2c_9-2c_{12}\right)a_{23}+2c_{10}-2c_7a_{13}+2c_8a_{24}\right)d_{a_{14}}, \left(2c_0\left(a_{13}a_{24}-a_{14}a_{23}\right)a_{24}-2c_4a_{23}a_{24}+\left(2c_5+2c_{12}\right)a_{24}+2c_1a_{14}a_{24}-2c_2\left(a_{13}a_{24}-a_{14}a_{23}\right)-2c_2a_{13}a_{24}+2c_3a_{24}^2-2c_7a_{14}+2c_{11}a_{13}-2c_{12}a_{24}+2c_{13}a_{23}-2c_{14}\right)d_{a_{23}}+\left(-2c_0\left(a_{13}a_{24}-a_{14}a_{23}\right)a_{23}+2c_4a_{23}^2-\left(2c_5+2c_{12}\right)a_{23}+2c_1\left(a_{13}a_{24}-a_{14}a_{23}\right)-2c_1a_{14}a_{23}+2c_2a_{13}a_{23}-2c_3a_{24}a_{23}+2c_6a_{14}-\left(2c_9-2c_{12}\right)a_{23}+2c_{10}-2c_7a_{13}+2c_8a_{24}\right)d_{a_{24}}, \left(-2c_0\left(a_{13}a_{24}-a_{14}a_{23}\right)a_{14}-2c_4\left(a_{13}a_{24}-a_{14}a_{23}\right)+2c_4a_{23}a_{14}-\left(2c_5+2c_{12}\right)a_{14}-2c_1a_{14}^2+2c_2a_{13}a_{14}-2c_3a_{24}a_{14}-\left(2c_9\right.\right.\right. \quad \textbf{(11.7)}$$

$$- 2 c_{12}) a_{14} + 2 c_{13} a_{13} - 2 c_{16} a_{24} + 2 c_{18} a_{23} - 2 c_{19}) d_{a_{13}} + (2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{13} - 2 c_4 a_{23} a_{13} + (2 c_5 + 2 c_{12}) a_{13} + 2 c_1 a_{14} a_{13} - 2 c_2 a_{13}^2 + 2 c_3 (a_{13} a_{24} - a_{14} a_{23}) + 2 c_3 a_{24} a_{13} + 2 c_8 a_{14} - 2 c_{12} a_{13} + 2 c_{15} a_{24} - 2 c_{16} a_{23} + 2 c_{17}) d_{a_{14}}, (-2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{14} - 2 c_4 (a_{13} a_{24} - a_{14} a_{23}) + 2 c_4 a_{23} a_{14} - (2 c_5 + 2 c_{12}) a_{14} - 2 c_1 a_{14}^2 + 2 c_2 a_{13} a_{14} - 2 c_3 a_{24} a_{14} - (2 c_9 - 2 c_{12}) a_{14} + 2 c_{13} a_{13} - 2 c_{16} a_{24} + 2 c_{18} a_{23} - 2 c_{19}) d_{a_{23}} + (2 c_0 (a_{13} a_{24} - a_{14} a_{23}) a_{13} - 2 c_4 a_{23} a_{13} + (2 c_5 + 2 c_{12}) a_{13} + 2 c_1 a_{14} a_{13} - 2 c_2 a_{13}^2 + 2 c_3 (a_{13} a_{24} - a_{14} a_{23}) + 2 c_3 a_{24} a_{13} + 2 c_8 a_{14} - 2 c_{12} a_{13} + 2 c_{15} a_{24} - 2 c_{16} a_{23} + 2 c_{17}) d_{a_{24}}]$$

For the integrability condition to be satisfied by affgQ, the following differential forms should vanish on the quadric defined by affgQ:

```
> [seq(seq(Mul(Mul(Diff(affgQ, alg), Diff(alphas[i],alg), alg),
  alphas[j],alg),j=1..nops(alphas)),i=1..nops(alphas))]:
```

So we take the remainders modulo affgQ:

```
> map(eq->normal(eq/convert(alg,`*`)), %):
> Groebner[NormalForm](%, [affgQ], tdeg(op(affineVars))):
```

And we obtain polynomials in the affineVars whose coefficients should be zero.

```
> map(collect, %, affineVars, 'distributed'): map(coeffs, %,
  affineVars):
> gensIntegrability_base := map(numer@normal, %):
```

The integrability equations do not depend on c[12].

```
> nops(select(has, gensIntegrability_base, c[12]));
```

$$0 \tag{11.8}$$

And we do the same for all the other charts. Equivalently, we permute the coordinates of C^4 and this induces a transformations of the c's that we apply to the equations above.

```
> permutations := map[3](zip, `=`, [0,1,2,3], combinat[permute]([0,
  1,2,3]));
```

$$permutations := [[0 = 0, 1 = 1, 2 = 2, 3 = 3], [0 = 0, 1 = 1, 2 = 3, 3 = 2], [0 = 0, 1 = 2, 2 = 1, 3 = 3], [0 = 0, 1 = 2, 2 = 3, 3 = 1], [0 = 0, 1 = 3, 2 = 1, 3 = 2], [0 = 0, 1 = 3, 2 = 2, 3 = 1], [0 = 1, 1 = 0, 2 = 2, 3 = 3], [0 = 1, 1 = 0, 2 = 3, 3 = 2], [0 = 1, 1 = 2, 2 = 0, 3 = 3], [0 = 1, 1 = 2, 2 = 3, 3 = 0], [0 = 1, 1 = 3, 2 = 0, 3 = 2], [0 = 1, 1 = 3, 2 = 2, 3 = 0], [0 = 2, 1 = 0, 2 = 1, 3 = 3], [0 = 2, 1 = 0, 2 = 3, 3 = 1], [0 = 2, 1 = 1, 2 = 0, 3 = 3], [0 = 2, 1 = 1, 2 = 3, 3 = 0], [0 = 2, 1 = 3, 2 = 0, 3 = 1], [0 = 2, 1 = 3, 2 = 1, 3 = 0], [0 = 3, 1 = 0, 2 = 1, 3 = 2], [0 = 3, 1 = 0, 2 = 2, 3 = 1], [0 = 3, 1 = 1, 2 \tag{11.9}$$

$= 0, 3 = 2], [0 = 3, 1 = 1, 2 = 2, 3 = 0], [0 = 3, 1 = 2, 2 = 0, 3 = 1], [0 = 3, 1 = 2, 2 = 1, 3 = 0]]$

Permutations of the coordinates of C^4 induces transformations of the Plucker coordinates:

```
> subs([seq(seq(p[i,j]=-p[j,i], j=0..i), i=0..3)], map(p -> zip(`=
  `, pluckerVars, subs(p, pluckerVars)), permutations));
```

$$[[p_{0,1}=p_{0,1}, p_{0,2}=p_{0,2}, p_{0,3}=p_{0,3}, p_{1,2}=p_{1,2}, p_{1,3}=p_{1,3}, p_{2,3}=p_{2,3}], [p_{0,1}=p_{0,1}, p_{0,2}=p_{0,3}, p_{0,3}=p_{0,2}, p_{1,2}=p_{1,3}, p_{1,3}=p_{1,2}, p_{2,3}=-p_{2,3}], [p_{0,1}=p_{0,2}, p_{0,2}=p_{0,1}, p_{0,3}=p_{0,3}, p_{1,2}=-p_{1,2}, p_{1,3}=p_{2,3}, p_{2,3}=p_{1,3}], [p_{0,1}=p_{0,2}, p_{0,2}=p_{0,3}, p_{0,3}=p_{0,1}, p_{1,2}=p_{2,3}, p_{1,3}=-p_{1,2}, p_{2,3}=-p_{1,3}], [p_{0,1}=p_{0,3}, p_{0,2}=p_{0,1}, p_{0,3}=p_{0,2}, p_{1,2}=-p_{1,3}, p_{1,3}=-p_{2,3}, p_{2,3}=p_{1,2}], [p_{0,1}=p_{0,3}, p_{0,2}=p_{0,2}, p_{0,3}=p_{0,1}, p_{1,2}=-p_{2,3}, p_{1,3}=-p_{1,3}, p_{2,3}=-p_{1,2}], [p_{0,1}=-p_{0,1}, p_{0,2}=p_{1,2}, p_{0,3}=p_{1,3}, p_{1,2}=p_{0,2}, p_{1,3}=p_{0,3}, p_{2,3}=p_{2,3}], [p_{0,1}=-p_{0,1}, p_{0,2}=p_{1,3}, p_{0,3}=p_{1,2}, p_{1,2}=p_{0,3}, p_{1,3}=p_{0,2}, p_{2,3}=-p_{2,3}], [p_{0,1}=p_{1,2}, p_{0,2}=-p_{0,1}, p_{0,3}=p_{1,3}, p_{1,2}=-p_{0,2}, p_{1,3}=p_{2,3}, p_{2,3}=p_{0,3}], [p_{0,1}=p_{1,2}, p_{0,2}=p_{1,3}, p_{0,3}=-p_{0,1}, p_{1,2}=p_{2,3}, p_{1,3}=-p_{0,2}, p_{2,3}=-p_{0,3}], [p_{0,1}=p_{1,3}, p_{0,2}=-p_{0,1}, p_{0,3}=p_{1,2}, p_{1,2}=-p_{0,3}, p_{1,3}=-p_{2,3}, p_{2,3}=p_{0,2}], [p_{0,1}=p_{1,3}, p_{0,2}=p_{1,2}, p_{0,3}=-p_{0,1}, p_{1,2}=-p_{2,3}, p_{1,3}=-p_{0,3}, p_{2,3}=-p_{0,2}], [p_{0,1}=-p_{0,2}, p_{0,2}=-p_{1,2}, p_{0,3}=p_{2,3}, p_{1,2}=p_{0,1}, p_{1,3}=p_{0,3}, p_{2,3}=p_{1,3}], [p_{0,1}=-p_{0,2}, p_{0,2}=p_{2,3}, p_{0,3}=-p_{1,2}, p_{1,2}=p_{0,3}, p_{1,3}=p_{0,1}, p_{2,3}=-p_{1,3}], [p_{0,1}=-p_{1,2}, p_{0,2}=-p_{0,2}, p_{0,3}=p_{2,3}, p_{1,2}=-p_{0,1}, p_{1,3}=p_{1,3}, p_{2,3}=p_{0,3}], [p_{0,1}=-p_{1,2}, p_{0,2}=p_{2,3}, p_{0,3}=-p_{0,2}, p_{1,2}=p_{1,3}, p_{1,3}=-p_{0,1}, p_{2,3}=-p_{0,3}], [p_{0,1}=p_{2,3}, p_{0,2}=-p_{0,2}, p_{0,3}=-p_{1,2}, p_{1,2}=-p_{0,3}, p_{1,3}=-p_{1,3}, p_{2,3}=p_{0,1}], [p_{0,1}=p_{2,3}, p_{0,2}=-p_{1,2}, p_{0,3}=-p_{0,2}, p_{1,2}=-p_{1,3}, p_{1,3}=-p_{0,3}, p_{2,3}=-p_{0,1}], [p_{0,1}=-p_{0,3}, p_{0,2}=-p_{1,3}, p_{0,3}=-p_{2,3}, p_{1,2}=p_{0,1}, p_{1,3}=p_{0,2}, p_{2,3}=p_{1,2}], [p_{0,1}=-p_{0,3}, p_{0,2}=-p_{2,3}, p_{0,3}=-p_{1,3}, p_{1,2}=p_{0,2}, p_{1,3}=p_{0,1}, p_{2,3}=-p_{1,2}], [p_{0,1}=-p_{1,3}, p_{0,2}=-p_{0,3}, p_{0,3}=-p_{2,3}, p_{1,2}=-p_{0,1}, p_{1,3}=p_{1,2}, p_{2,3}=p_{0,2}], [p_{0,1}=-p_{1,3}, p_{0,2}=-p_{2,3}, p_{0,3}=-p_{0,3}, p_{1,2}=p_{1,2}, p_{1,3}=-p_{0,1}, p_{2,3}=-p_{0,2}], [p_{0,1}=-p_{2,3}, p_{0,2}=-p_{0,3}, p_{0,3}=-p_{1,3}, p_{1,2}=-p_{0,2}, p_{1,3}=-p_{1,2}, p_{2,3}=p_{0,1}], [p_{0,1}=-p_{2,3}, p_{0,2}=-p_{1,3}, p_{0,3}=-p_{0,3}, p_{1,2}=-p_{1,2}, p_{1,3}=-p_{0,2}, p_{2,3}=-p_{0,1}]] \tag{11.10}$$

```
> subs(c=d, map(subs, %, gQ)):
> map( p->subs(d=c,solve([coeffs(gQ - p, pluckerVars)], cvars)[1]),
  %):
> map(op, map(subs, %, gensIntegrability_base)):
> gensIntegrability := [op(gbCoiso), op(convert(map(normal, %),
```

```
set))]:
> gbIntegrability := fgb_gbasis(gensIntegrability, 0, [], cvars0):
```

Now we want to check Proposition 3 of “Computing ...”

```
> eqsChowFormOrSquare := [seq((1-t)*p, p in gbChowForm), seq(t*p, p
  in gbSquare )]:
```

We compute a Groebner basis for the triple intersection of P_ChowConic, P_ChowLines and P_Square.

```
> gbChowFormOrSquare := fgb_gbasis_elim(eqsChowFormOrSquare, 0,
  [t], cvars0):
> nops(gbChowFormOrSquare);
```

$$466 \tag{11.11}$$

The integrability ideal is contained in the triple intersection.

```
> map(Groebner[Reduce], gbIntegrability, gbChowFormOrSquare, tdeg
  (op(cvars0)));
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,$$
$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0] \tag{11.12}$$

We now compare the Hilbert series of both ideals.

```
> fgb_hilbert(gbIntegrability, 0, cvars0, [], t): hilbIntegrability
  := %[1]/(1-t)^%[2];
```

$$\mathit{hilbIntegrability} := \frac{1}{(1-t)^9}\left(t^{11} - 8\,t^{10} + 27\,t^9 - 49\,t^8 + 30\,t^7 + 28\,t^6 - 30\,t^5 + 78\,t^4 + 77\,t^3 + 66\,t^2 + 11\,t + 1\right) \tag{11.13}$$

```
> fgb_hilbert(gbChowFormOrSquare, 0, cvars0, [], t):
  hilbChowFormOrSquare := %[1]/(1-t)^%[2];
```

$$\mathit{hilbChowFormOrSquare} := \frac{1}{(1-t)^9}\left(t^{12} - 8\,t^{11} + 28\,t^{10} - 57\,t^9 + 77\,t^8 - 96\,t^7 + 112\,t^6 - 66\,t^5 + 87\,t^4 + 76\,t^3 + 66\,t^2 + 11\,t + 1\right) \tag{11.14}$$

There is exactly one polynomial (modulo the ideal Integrability) which is in the ideal

ChowFormOrSquare but not in Integrability and it is a cubic.

```
> normal(hilbIntegrability - hilbChowFormOrSquare);
```

$$t^3 \tag{11.15}$$

Let us find this polynomial:

```
> pol3 := (select(p -> degree(p) = 3, convert(gbChowFormOrSquare,
  set)) minus select(p -> degree(p) = 3, convert(gbIntegrability,
  set)))[1];
```

$$pol3 := -c_0 c_9 c_{20} + c_1 c_4 c_{20} + 3 c_1 c_{10} c_{18} - 2 c_1 c_{13} c_{17} + c_2 c_3 c_{20} + 2 c_2 c_8 c_{19} - 4 c_2 c_{10} c_{16} - c_3 c_{11} c_{17} - 2 c_4 c_6 c_{19} + 4 c_4 c_7 c_{17} - c_5 c_6 c_{18} + c_5 c_{11} c_{15} - c_7 c_8 c_{18} - c_8 c_{11} c_{16} + c_9 c_{11} c_{15} \tag{11.16}$$

```
> Groebner[Reduce](pol3, gbChowFormOrSquare, tdeg(op(cvars0)));
```

$$0 \tag{11.17}$$

```
> Groebner[Reduce](pol3, gbIntegrability, tdeg(op(cvars0)));
```

$$-c_0 c_9 c_{20} + c_1 c_4 c_{20} + 3 c_1 c_{10} c_{18} - 2 c_1 c_{13} c_{17} + c_2 c_3 c_{20} + 2 c_2 c_8 c_{19} - 4 c_2 c_{10} c_{16} - c_3 c_{11} c_{17} - 2 c_4 c_6 c_{19} + 4 c_4 c_7 c_{17} - c_5 c_6 c_{18} + c_5 c_{11} c_{15} - c_7 c_8 c_{18} - c_8 c_{11} c_{16} + c_9 c_{11} c_{15} \tag{11.18}$$

```
> Groebner[Reduce](map(`*`, cvars0, pol3), gbIntegrability, tdeg(op
  (cvars0)));
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0] \tag{11.19}$$

This concludes the proof of Proposition 3.

[Annex] Save to file

Save to file the Groebner bases for future use.

```
> GBs := [gbCoiso, gbChowLines, gbChowConic, gbHurwitz, gbSquare,
  gbIntegrability]:
> fd := fopen("GBs.mpl", WRITE):
> fprintf(fd, "gbCoiso := %a:\n\ngbChowLines := %a:\n\ngbChowConic
  := %a:\n\ngbHurwitz := %a:\n\ngbSquare := %a:\n\ngbIntegrability
  := %a:\n\n", op(GBs));
```

$$1242441 \tag{12.1}$$

```
> fclose(fd):
```

Save the Groebner bases for use in Magma.

```
> fd := fopen("GBs.magma", WRITE):
> fprintf(fd, "ICoiso := Ideal( %a );\n\nIChowLines := Ideal( %a );
  \n\nIChowConic := Ideal( %a );\n\nIHurwitz := Ideal( %a );
  \n\nISquare := Ideal( %a );\n\nJ := Ideal( %a );\n\n",
  op(subs([seq(c[i]=c||i, i=0..20)], GBs)));
```

$$840293 \tag{12.2}$$

```
> fclose(fd):
```